

# Coupled Similarity and Contrasimilarity, and How to Compute Them

Benjamin Bisping\*    Luisa Montanari

April 18, 2024

## Abstract

This theory surveys and extends characterizations of *coupled similarity* and *contrasimilarity*, and proves properties relevant for algorithms computing their simulation preorders and equivalences.

Coupled similarity and contrasimilarity are two weak forms of bisimilarity for systems with internal behavior. They have outstanding applications in contexts where internal choices must transparently be distributed in time or space, for example, in process calculi encodings or in action refinements.

Our key contribution is to characterize the coupled simulation and contrasimulation preorders by *reachability games*. We also show how preexisting definitions coincide and that they can be reformulated using *coupled delay simulations*. We moreover verify a polynomial-time coinductive fixed-point algorithm computing the coupled simulation preorder. Through reduction proofs, we establish that deciding coupled similarity is at least as complex as computing weak similarity; and that contrasimilarity checking is at least as hard as trace inclusion checking.

## Contents

<b>1</b>	<b>Introduction</b>	<b>3</b>
1.1	This Theory . . . . .	3
1.2	Coupled Similarity vs. Weak Bisimilarity vs. Contrasimilarity in a Nutshell . . . . .	4
1.3	Modal Intuition . . . . .	4
<b>2</b>	<b>Preliminaries</b>	<b>4</b>
2.1	Labeled Transition Systems . . . . .	4
2.2	Transition Systems with Silent Steps . . . . .	6
2.3	Finite Transition Systems with Silent Steps . . . . .	12
2.4	Simple Games . . . . .	12
<b>3</b>	<b>Notions of Equivalence</b>	<b>14</b>
3.1	Strong Simulation and Bisimulation . . . . .	14
3.2	Weak Simulation . . . . .	15
3.3	Weak Bisimulation . . . . .	17
3.4	Trace Inclusion . . . . .	19
3.5	Delay Simulation . . . . .	19
3.6	Coupled Equivalences . . . . .	19

---

\*TU Berlin, Germany, <https://bbisping.de>, [benjamin.bisping@tu-berlin.de](mailto:benjamin.bisping@tu-berlin.de).

<b>4</b>	<b>Contrasimulation</b>	<b>20</b>
4.1	Definition of Contrasimulation . . . . .	20
4.2	Intermediate Relation Mimicking Contrasim . . . . .	22
4.3	Over-Approximating Contrasimulation by a Single-Step Version . . . . .	22
<b>5</b>	<b>Coupled Simulation</b>	<b>23</b>
5.1	Van Glabbeeks’s Coupled Simulation . . . . .	23
5.2	Position between Weak Simulation and Weak Bisimulation . . . . .	23
5.3	Coupled Simulation and Silent Steps . . . . .	24
5.4	Closure, Preorder and Symmetry Properties . . . . .	25
5.5	Coinductive Coupled Simulation Preorder . . . . .	25
5.6	Coupled Simulation Join . . . . .	26
5.7	Coupled Delay Simulation . . . . .	26
5.8	Relationship to Contrasimulation and Weak Simulation . . . . .	27
5.9	$\tau$ -Reachability (and Divergence) . . . . .	27
5.10	On the Connection to Weak Bisimulation . . . . .	29
5.11	Reduction Semantics Coupled Simulation . . . . .	30
5.12	Coupled Simulation as Two Simulations . . . . .	30
5.13	S-coupled Simulation . . . . .	31
<b>6</b>	<b>Game for Coupled Similarity with Delay Formulation</b>	<b>34</b>
6.1	The Coupled Simulation Preorder Game Using Delay Steps . . . . .	34
6.2	Coupled Simulation Implies Winning Strategy . . . . .	35
6.3	Winning Strategy Induces Coupled Simulation . . . . .	36
<b>7</b>	<b>Fixed Point Algorithm for Coupled Similarity</b>	<b>36</b>
7.1	The Algorithm . . . . .	36
7.2	Correctness . . . . .	36
<b>8</b>	<b>The Contrasimulation Preorder Word Game</b>	<b>37</b>
8.1	Contrasimulation Implies Winning Strategy in Word Game (Completeness) . . . . .	38
8.2	Winning Strategy Implies Contrasimulation in Word Game (Soundness) . . . . .	39
<b>9</b>	<b>The Contrasimulation Preorder Set Game</b>	<b>39</b>
9.1	Contrasimulation Implies Winning Strategy in Set Game (Completeness) . . . . .	40
9.2	Winning Strategy Implies Contrasimulation in Set Game (Soundness) . . . . .	41
<b>10</b>	<b>Infinitary Hennessy–Milner Logic</b>	<b>42</b>
10.1	Satisfaction Relation . . . . .	43
10.2	Distinguishing Formulas . . . . .	43
10.3	Weak-NOR Hennessy–Milner Logic . . . . .	44
<b>11</b>	<b>Weak HML and the Contrasimulation Set Game</b>	<b>44</b>
11.1	Distinguishing Formulas at Winning Attacker Positions . . . . .	45
11.2	Attacker Wins on Pairs with Distinguishing Formulas . . . . .	46
<b>12</b>	<b>Reductions and <math>\tau</math>-sinks</b>	<b>47</b>
12.1	$\tau$ -Sink Properties . . . . .	48
12.2	Contrasimulation Equals Weak Simulation on $\tau$ -Sink Systems . . . . .	48
12.3	Contrasimulation Equals Weak Trace Inclusion on $\tau$ -Sink Systems . . . . .	48
12.4	Weak Simulation Invariant under $\tau$ -Sink Extension . . . . .	49
12.5	Trace Inclusion Invariant under $\tau$ -Sink Extension . . . . .	50
	<b>References</b>	<b>50</b>




Figure 1: Hierarchy of weak behavioral preorders/equivalences. Arrows denote implication of preordering. Thinner arrows collapse into bi-implication for systems without internal steps. Blue parts indicate a slope of decision problem complexities.

## 1 Introduction

Coupled similarity and contrasimilarity are among the weakest abstractions of bisimilarity for systems with silent steps presented in van Glabbeek’s *linear-time-branching-time spectrum* of behavioral equivalences [8]. In particular, they are *weaker than weak bisimilarity* in that they impose a weaker form of symmetry on the bisimulation link between compared states; coupled similarity implies contrasimilarity. They are weak bisimilarities, however, in the sense that, on systems with no internal behavior, they coincide with strong bisimilarity.

### 1.1 This Theory

This theory contains the Isabelle/HOL formalization for two related lines of publication, which present the first algorithms to check coupled similarity and contrasimilarity for pairs of states:

- *Computing coupled similarity*: Bisping and Nestmann’s TACAS 2019 paper [4] and Bisping’s master thesis [1] establish the first decision procedures for coupled similarity checking. This is done through a game-based approach. Also, the work introduces the idea that  $\tau$ -sinks can be used to reduce from weak simulation preorder to coupled simulation preorder.
- *Game characterization of contrasimilarity*: Bisping and Montanari’s EXPRESS/SOS 2021 paper [3] and Montanari’s bachelor thesis provide the first game characterization of contrasimilarity. The present Isabelle theory extends this work by also showing a reduction from weak trace preorder to contrasimulation preorder and by linking the game to a modal characterization of contrasimilarity.

Combined, the results establish a slope of complexity between weak bisimilarity, coupled similarity, and contrasimilarity with the equivalence problems becoming harder for coarser equivalences. See Figure 1 for a graphical representation.

## 1.2 Coupled Similarity vs. Weak Bisimilarity vs. Contrasimilarity in a Nutshell

In coupled simulation semantics, the CCS process  $\tau.a + \tau.(\tau.b + \tau.c)$  with gradual internal choice equals  $\tau.a + \tau.b + \tau.c$ , which has just one internal choice point. In weak bisimulation semantics, this equality does not hold as the intermediate choice point  $\tau.b + \tau.c$  of the first process does not match symmetrically to any state of the second process.

The equality also holds in contrasimulation semantics. Contrasimulation moreover blurs the lines between non-determinism of visible behavior and internal non-deterministic choice by considering  $a.b + a.c$  to be indistinguishable from  $a.(\tau.b + \tau.c)$ . This equality does not hold under coupled similarity. Therefore, contrasimilarity is strictly coarser than coupled similarity.

For a more detailed exposition about the nuances of coupled similarity and contrasimilarity we refer to our publications [4, 5, 3].

## 1.3 Modal Intuition

The modal characterization of contrasimilarity at the end of this theory gives a nice intuition for why contrasimilarity is a sensible weakening for bisimilarity. We show that the following Hennessy–Milner logic (with  $\langle \varepsilon \rangle$  denoting places of possible internal behavior) characterizes contrasimilarity.

$$\varphi ::= \langle \varepsilon \rangle \langle a \rangle \varphi \quad | \quad \langle \varepsilon \rangle \bigwedge_{i \in I} \neg \varphi_i \quad (\text{with } a \neq \tau).$$

It is a “nice” abstraction of strong bisimilarity, since it can be obtained from the following complete fragment of Hennessy–Milner logic by inserting places for unobservable behavior in front of each constructor.

$$\varphi ::= \langle a \rangle \varphi \quad | \quad \bigwedge_{i \in I} \neg \varphi_i.$$

This modal formulation is important for a unified algorithmic treatment of weak behavioral equivalences in [2].

## 2 Preliminaries

### 2.1 Labeled Transition Systems

```

theory Transition_Systems
  imports Main
begin

locale lts =
fixes
  trans :: <'s ⇒ 'a ⇒ 's ⇒ bool> ("_ ↦ _" [70, 70, 70] 80)

begin

abbreviation step_pred :: <'s ⇒ ('a ⇒ bool) ⇒ 's ⇒ bool>
  where
    <step_pred p af q ≡ ∃ a. af a ∧ trans p a q>

inductive steps :: <'s ⇒ ('a ⇒ bool) ⇒ 's ⇒ bool>
  ("_ ↦* _" [70, 70, 70] 80)
where
  refl: <p ↦* A p> | step: <p ↦* A q1 ⇒ q1 ↦ a q ⇒ A a ⇒ (p ↦* A q)>

lemma steps_one_step:

```

```

assumes
  <p  $\mapsto$ a p' >
  <A a >
shows
  <p  $\mapsto$ * A p' > <proof>

lemma steps_concat:
assumes
  <p'  $\mapsto$ * A p'' >
  <p  $\mapsto$ * A p' >
shows
  <p  $\mapsto$ * A p'' > <proof>

lemma steps_left:
assumes
  <p  $\neq$  p' >
  <p  $\mapsto$ * A p' >
shows
  < $\exists$ p'' a . p  $\mapsto$ a p''  $\wedge$  A a  $\wedge$  p''  $\mapsto$ * A p' >
  <proof>

lemma steps_no_step:
assumes
  < $\wedge$  a p' . p  $\mapsto$ a p'  $\implies$   $\neg$ A a >
  <p  $\neq$  p'' >
  <p  $\mapsto$ * A p'' >
shows
  <False >
  <proof>

lemma steps_no_step_pos:
assumes
  < $\wedge$  a p' . p  $\mapsto$ a p'  $\implies$   $\neg$ A a >
  <p  $\mapsto$ * A p' >
shows
  <p = p' >
  <proof>

lemma steps_loop:
assumes
  < $\wedge$  a p' . p  $\mapsto$ a p'  $\implies$  p = p' >
  <p  $\neq$  p'' >
  <p  $\mapsto$ * A p'' >
shows
  <False >
  <proof>

corollary steps_transp:
  <transp ( $\lambda$  p p'. p  $\mapsto$ * A p') >
  <proof>

lemma steps_spec:
assumes
  <p  $\mapsto$ * A' p' >
  < $\wedge$  a . A' a  $\implies$  A a >
shows
  <p  $\mapsto$ * A p' > <proof>

```

```
interpretation preorder <( $\lambda$  p p'. p  $\mapsto^*$  A p')> < $\lambda$  p p'. p  $\mapsto^*$  A p'  $\wedge$   $\neg$ (p'  $\mapsto^*$  A p)>
  <proof>
```

If one can reach only a finite portion of the graph following  $\mapsto^*$  A, and all cycles are loops, then there must be nodes which are maximal wrt.  $\mapsto^*$  A.

```
lemma step_max_deadlock:
  fixes A q
  assumes
    antiysmm: < $\bigwedge$  r1 r2. r1  $\mapsto^*$  A r2  $\wedge$  r2  $\mapsto^*$  A r1  $\implies$  r1 = r2> and
    finite: <finite {q'. q  $\mapsto^*$  A q'}> and
    no_max: < $\forall$  q'. q  $\mapsto^*$  A q'  $\implies$  ( $\exists$ q''. q'  $\mapsto^*$  A q''  $\wedge$  q'  $\neq$  q'')>
  shows
    False
  <proof>
```

end — end of lts

```
lemma lts_impl_steps2:
  assumes
    <lts.steps step1 p1 ap p2>
    < $\bigwedge$  p1 a p2 . step1 p1 a p2  $\wedge$  P p1 a p2  $\implies$  step2 p1 a p2>
    < $\bigwedge$  p1 a p2 . P p1 a p2>
  shows
    <lts.steps step2 p1 ap p2>
  <proof>
```

```
lemma lts_impl_steps:
  assumes
    <lts.steps step1 p1 ap p2>
    < $\bigwedge$  p1 a p2 . step1 p1 a p2  $\implies$  step2 p1 a p2>
  shows
    <lts.steps step2 p1 ap p2>
  <proof>
```

end

## 2.2 Transition Systems with Silent Steps

```
theory Weak_Transition_Systems
  imports Transition_Systems
begin

locale lts_tau = lts trans for
  trans :: <'s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool> ("_  $\mapsto$ _ _" [70, 70, 70] 80) + fixes
   $\tau$  :: <'a> begin

definition tau :: <'a  $\Rightarrow$  bool> where <tau a  $\equiv$  (a =  $\tau$ )>

lemma tau_tau[simp]: <tau  $\tau$ > <proof>

abbreviation weak_step :: <'s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool>
  ("_  $\Rightarrow$ _ _" [70, 70, 70] 80)
where
  <(p  $\Rightarrow$ a q)  $\equiv$  ( $\exists$  pq1 pq2.
    p  $\mapsto^*$  tau pq1  $\wedge$ 
    pq1  $\mapsto$ a pq2  $\wedge$ 
```

```

    pq2  $\mapsto^*$  tau q) >

lemma step_weak_step:
  assumes <p  $\mapsto^a$  p' >
  shows <p  $\Rightarrow^a$  p' >
  <proof>

abbreviation weak_step_tau :: <'s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool >
  ("  $\Rightarrow^a$  _ _" [70, 70, 70] 80)
where
  <(p  $\Rightarrow^a$  q)  $\equiv$ 
    (tau a  $\longrightarrow$  p  $\mapsto^*$  tau q)  $\wedge$ 
    ( $\neg$ tau a  $\longrightarrow$  p  $\Rightarrow^a$  q) >

abbreviation weak_step_delay :: <'s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool >
  ("  $\Rightarrow^d$  _ _" [70, 70, 70] 80)
where
  <(p  $\Rightarrow^d$  a q)  $\equiv$ 
    (tau a  $\longrightarrow$  p  $\mapsto^*$  tau q)  $\wedge$ 
    ( $\neg$ tau a  $\longrightarrow$  ( $\exists$  pq.
      p  $\mapsto^*$  tau pq  $\wedge$ 
      pq  $\mapsto^a$  q)) >

lemma weak_step_delay_implies_weak_tau:
  assumes <p  $\Rightarrow^d$  a p' >
  shows <p  $\Rightarrow^a$  p' >
  <proof>

lemma weak_step_delay_left:
  assumes
    < $\neg$  p0  $\mapsto^a$  p1 >
    <p0  $\Rightarrow^d$  a p1 >
    < $\neg$ tau a >
  shows
    < $\exists$  p0' t. tau t  $\wedge$  p0  $\mapsto^t$  p0'  $\wedge$  p0'  $\Rightarrow^d$  a p1 >
  <proof>

primrec weak_step_seq :: <'s  $\Rightarrow$  'a list  $\Rightarrow$  's  $\Rightarrow$  bool >
  ("  $\Rightarrow^s$  _ _" [70, 70, 70] 80)
  where
    <weak_step_seq p0 [] p1 = p0  $\mapsto^*$  tau p1 >
    | <weak_step_seq p0 (a#A) p1 = ( $\exists$  p01 . p0  $\Rightarrow^a$  p01  $\wedge$  weak_step_seq p01 A p1) >

lemma step_weak_step_tau:
  assumes <p  $\mapsto^a$  p' >
  shows <p  $\Rightarrow^a$  p' >
  <proof>

lemma step_tau_refl:
  shows <p  $\Rightarrow^a$  p >
  <proof>

lemma weak_step_tau_weak_step[simp]:
  assumes <p  $\Rightarrow^a$  p' > < $\neg$  tau a >
  shows <p  $\Rightarrow^a$  p' >
  <proof>

```

```

lemma weak_steps:
  assumes
    <p  $\Rightarrow_a$  p'>
    < $\bigwedge$  a . tau a  $\implies$  A a>
    <A a>
  shows
    <p  $\mapsto^*$  A p'>
  <proof>

lemma weak_step_impl_weak_tau:
  assumes
    <p  $\Rightarrow_a$  p'>
  shows
    <p  $\Rightarrow^{\sim} a$  p'>
  <proof>

lemma weak_impl_strong_step:
  assumes
    <p  $\Rightarrow_a$  p''>
  shows
    <( $\exists$  a' p' . tau a'  $\wedge$  p  $\mapsto_{a'}$  p')  $\vee$  ( $\exists$  p' . p  $\mapsto_a$  p')>
  <proof>

lemma weak_step_extend:
  assumes
    <p1  $\mapsto^*$  tau p2>
    <p2  $\Rightarrow^{\sim} a$  p3>
    <p3  $\mapsto^*$  tau p4>
  shows
    <p1  $\Rightarrow^{\sim} a$  p4>
  <proof>

lemma weak_step_tau_tau:
  assumes
    <p1  $\mapsto^*$  tau p2>
    <tau a>
  shows
    <p1  $\Rightarrow^{\sim} a$  p2>
  <proof>

lemma weak_single_step[iff]:
  <p  $\Rightarrow_{[a]}$  p'  $\longleftrightarrow$  p  $\Rightarrow^{\sim} a$  p'>
  <proof>

abbreviation weak_enabled :: <'s  $\Rightarrow$  'a  $\Rightarrow$  bool> where
  <weak_enabled p a  $\equiv$ 
   $\exists$  pq1 pq2. p  $\mapsto^*$  tau pq1  $\wedge$  pq1  $\mapsto_a$  pq2>

lemma weak_enabled_step:
  shows <weak_enabled p a = ( $\exists$  p' . p  $\Rightarrow_a$  p')>
  <proof>

lemma step_tau_concat:
  assumes
    <q  $\Rightarrow^{\sim} a$  q'>
    <q'  $\Rightarrow^{\sim} \tau$  q1>
  shows <q  $\Rightarrow^{\sim} a$  q1>

```



```

<proof>

lemma tau_step_concat:
  assumes
    <math>q \Rightarrow^{\tau} q'>
    <math>q' \Rightarrow^a q1>
  shows <math>q \Rightarrow^a q1>
<proof>

lemma tau_word_concat:
  assumes
    <math>q \Rightarrow^{\tau} q'>
    <math>q' \Rightarrow^{\$A} q1>
  shows <math>q \Rightarrow^{\$A} q1>
<proof>

lemma strong_weak_transition_system:
  assumes
    <math>\bigwedge p q a. p \mapsto a q \implies \neg \text{tau } a>
    <math>\neg \text{tau } a>
  shows
    <math>p \Rightarrow^a p' = p \mapsto a p'>
<proof>

lemma rev_seq_split :
  assumes <math>q \Rightarrow^{\$(xs @ [x])} q1>
  shows <math>\exists q'. q \Rightarrow^{\$xs} q' \wedge q' \Rightarrow^x q1>
<proof>

lemma rev_seq_concat:
  assumes
    <math>q \Rightarrow^{\$as} q'>
    <math>q' \Rightarrow^{\$A} q1>
  shows <math>q \Rightarrow^{\$(as@A)} q1>
<proof>

lemma rev_seq_step_concat :
  assumes
    <math>q \Rightarrow^{\$as} q'>
    <math>q' \Rightarrow^a q1>
  shows <math>q \Rightarrow^{\$(as@[a])} q1>
<proof>

lemma rev_seq_dstep_concat :
  assumes
    <math>q \Rightarrow^{\$as} q'>
    <math>q' \Rightarrow^{\triangleright a} q1>
  shows <math>q \Rightarrow^{\$(as@[a])} q1>
<proof>

lemma word_tau_concat:
  assumes
    <math>q \Rightarrow^{\$A} q'>
    <math>q' \Rightarrow^{\tau} q1>
  shows <math>q \Rightarrow^{\$A} q1>
<proof>

```

```

lemma list_rev_split :
  assumes <A ≠ []>
  shows <∃ as a. A = as@[a]>
  <proof>

primrec taufree :: '<a list ⇒ 'a list>
  where
    <taufree [] = []>
  | <taufree (a#A) = (if tau a then taufree A else a#(taufree A))>

lemma weak_step_over_tau :
  assumes
    <p ⇒$A p'>
  shows <p ⇒$(taufree A) p'> <proof>

lemma app_tau_taufree_list :
  assumes
    <∀ a ∈ set A. ¬tau a>
    <b = τ>
  shows <A = taufree (A@[b])> <proof>

lemma word_steps_ignore_tau_addition:
  assumes
    <∀ a ∈ set A. a ≠ τ>
    <p ⇒$ A p'>
    <filter (λa. a ≠ τ) A' = A>
  shows
    <p ⇒$ A' p'>
  <proof>

lemma word_steps_ignore_tau_removal:
  assumes
    <p ⇒$ A p'>
  shows
    <p ⇒$ (filter (λa. a ≠ τ) A) p'>
  <proof>

definition weak_tau_succs :: "'s set ⇒ 's set" where
  <weak_tau_succs Q = {q1. ∃ q ∈ Q. q ⇒τ q1}>

definition dsuccs :: "'a ⇒ 's set ⇒ 's set" where
  <dsuccs a Q = {q1. ∃ q ∈ Q. q ⇒a q1}>

definition word_reachable_via_delay :: "'a list ⇒ 's ⇒ 's ⇒ 's ⇒ bool" where
  <word_reachable_via_delay A p p0 p1 = (∃ p00. p ⇒$(butlast A) p00 ∧ p00 ⇒(last A) p0 ∧ p0 ⇒τ p1)>

primrec dsuccs_seq_rec :: "'a list ⇒ 's set ⇒ 's set" where
  <dsuccs_seq_rec [] Q = Q> |
  <dsuccs_seq_rec (a#as) Q = dsuccs a (dsuccs_seq_rec as Q)>

lemma in_dsuccs_implies_word_reachable:
  assumes
    <q' ∈ dsuccs_seq_rec (rev A) {q}>
  shows
    <q ⇒$A q'>

```

```

⟨proof⟩

lemma word_reachable_implies_in_dsuccs :
  assumes
    ⟨q ⇒$A q'⟩
  shows ⟨q' ∈ weak_tau_succs (dsuccs_seq_rec (rev A) {q})⟩ ⟨proof⟩

lemma simp_dsuccs_seq_rev:
  assumes
    ⟨Q = dsuccs_seq_rec (rev A) {q0}⟩
  shows
    ⟨dsuccs a Q = dsuccs_seq_rec (rev (A@[a])) {q0}⟩
⟨proof⟩

abbreviation tau_max :: <'s ⇒ bool> where
  ⟨tau_max p ≡ (∀p'. p ⟶* tau p' ⟶ p = p')⟩

lemma tau_max_deadlock:
  fixes q
  assumes
    ⟨∧ r1 r2. r1 ⟶* tau r2 ∧ r2 ⟶* tau r1 ⟹ r1 = r2⟩ — contracted cycles (anti-symmetry)
    ⟨finite {q'. q ⟶* tau q'}⟩
  shows
    ⟨∃ q' . q ⟶* tau q' ∧ tau_max q'⟩
⟨proof⟩

abbreviation stable_state :: <'s ⇒ bool> where
  ⟨stable_state p ≡ # p' . step_pred p tau p'⟩

lemma stable_tauclosure_only_loop:
  assumes
    ⟨stable_state p⟩
  shows
    ⟨tau_max p⟩
⟨proof⟩

coinductive divergent_state :: <'s ⇒ bool> where
  omega: ⟨divergent_state p' ⟹ tau t ⟹ p ⟶t p' ⟹ divergent_state p⟩

lemma ex_divergent:
  assumes ⟨p ⟶a p⟩ ⟨tau a⟩
  shows ⟨divergent_state p⟩
⟨proof⟩

lemma ex_not_divergent:
  assumes ⟨∀ a q. p ⟶a q ⟶ ¬ tau a⟩ ⟨divergent_state p⟩
  shows ⟨False⟩ ⟨proof⟩

lemma perpetual_instability_divergence:
  assumes
    ⟨∀ p' . p ⟶* tau p' ⟶ ¬ stable_state p'⟩
  shows
    ⟨divergent_state p⟩
⟨proof⟩

corollary non_divergence_implies_eventual_stability:
  assumes

```

```

    <¬ divergent_state p>
  shows
    <∃ p' . p ⟶* tau p' ∧ stable_state p'>
  <proof>

end — context lts_tau

```

## 2.3 Finite Transition Systems with Silent Steps

```

locale lts_tau_finite = lts_tau trans  $\tau$  for
  trans :: <'s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool> and
   $\tau$  :: <'a> +
assumes
  finite_state_set: <finite (top::'s set)>
begin

lemma finite_state_rel: <finite (top::('s rel))>
  <proof>

end

end

```

## 2.4 Simple Games

```

theory Simple_Game
imports
  Main
begin

```

Simple games are games where player0 wins all infinite plays.

```

locale simple_game =
fixes
  game_move :: <'s  $\Rightarrow$  's  $\Rightarrow$  bool> ("_ ⟶ $\heartsuit$ _" [70, 70] 80) and
  player0_position :: <'s  $\Rightarrow$  bool>
begin

abbreviation player1_position :: <'s  $\Rightarrow$  bool>
  where <player1_position s  $\equiv$  ¬ player0_position s>

```

— Plays (to be precise: play prefixes) are lists. We model them with the most recent move at the beginning. (For our purpose it's enough to consider finite plays.)

```

type_synonym ('s2) play = <'s2 list>
type_synonym ('s2) strategy = <'s2 play  $\Rightarrow$  's2>
type_synonym ('s2) posstrategy = <'s2  $\Rightarrow$  's2>

definition strategy_from_positional :: <'s posstrategy  $\Rightarrow$  's strategy> where
  <strategy_from_positional pf = ( $\lambda$  play. pf (hd play))>

inductive_set plays :: <'s  $\Rightarrow$  's play set>
  for initial :: 's where
  <[initial]  $\in$  plays initial> |
  <p#play  $\in$  plays initial  $\Rightarrow$  p ⟶ $\heartsuit$  p'  $\Rightarrow$  p'#p#play  $\in$  plays initial>

definition play_continuation :: <'s play  $\Rightarrow$  's play  $\Rightarrow$  bool>
  where <play_continuation p1 p2  $\equiv$  (drop (length p2 - length p1) p2) = p1>

```

— Plays for a given player 0 strategy

```

inductive_set plays_for_0strategy :: <'s strategy  $\Rightarrow$  's  $\Rightarrow$  's play set>
  for f initial where
  init: <[initial]  $\in$  plays_for_0strategy f initial> |
  p0move:
    <n0#play  $\in$  plays_for_0strategy f initial  $\implies$  player0_position n0  $\implies$  n0  $\mapsto$   $\heartsuit$  f (n0#play)
     $\implies$  (f (n0#play))#n0#play  $\in$  plays_for_0strategy f initial> |
  p1move:
    <n1#play  $\in$  plays_for_0strategy f initial  $\implies$  player1_position n1  $\implies$  n1  $\mapsto$   $\heartsuit$  n1'
     $\implies$  n1'#n1#play  $\in$  plays_for_0strategy f initial>

```

**lemma** strategy0\_step:

```

assumes
  <n0 # n1 # rest  $\in$  plays_for_0strategy f initial>
  <player0_position n1>
shows
  <f (n1 # rest) = n0>
  <proof>

```

**inductive\_set** plays\_for\_1strategy :: <'s strategy  $\Rightarrow$  's  $\Rightarrow$  's play set>

```

for f initial where
  init: <[initial]  $\in$  plays_for_1strategy f initial> |
  p0move:
    <n0#play  $\in$  plays_for_1strategy f initial  $\implies$  player0_position n0  $\implies$  n0  $\mapsto$   $\heartsuit$  n0'
     $\implies$  n0'#n0#play  $\in$  plays_for_1strategy f initial> |
  p1move:
    <n1#play  $\in$  plays_for_1strategy f initial  $\implies$  player1_position n1  $\implies$  n1  $\mapsto$   $\heartsuit$  f (n1#play)
     $\implies$  (f (n1#play))#n1#play  $\in$  plays_for_1strategy f initial>

```

**definition** positional\_strategy :: <'s strategy  $\Rightarrow$  bool> **where**  
 <positional\_strategy f  $\equiv$   $\forall$ r1 r2 n. f (n # r1) = f (n # r2)>

A strategy is sound if it only decides on enabled transitions.

**definition** sound\_0strategy :: <'s strategy  $\Rightarrow$  's  $\Rightarrow$  bool> **where**

```

<sound_0strategy f initial  $\equiv$ 
   $\forall$  n0 play .
    n0#play  $\in$  plays_for_0strategy f initial  $\wedge$ 
    player0_position n0  $\rightarrow$  n0  $\mapsto$   $\heartsuit$  f (n0#play)>

```

**definition** sound\_1strategy :: <'s strategy  $\Rightarrow$  's  $\Rightarrow$  bool> **where**

```

<sound_1strategy f initial  $\equiv$ 
   $\forall$  n1 play .
    n1#play  $\in$  plays_for_1strategy f initial  $\wedge$ 
    player1_position n1  $\rightarrow$  n1  $\mapsto$   $\heartsuit$  f (n1#play)>

```

**lemma** strategy0\_plays\_subset:

```

assumes <play  $\in$  plays_for_0strategy f initial>
shows <play  $\in$  plays initial>
  <proof>

```

**lemma** strategy1\_plays\_subset:

```

assumes <play  $\in$  plays_for_1strategy f initial>
shows <play  $\in$  plays initial>
  <proof>

```

**lemma** no\_empty\_plays:

```

assumes <[]  $\in$  plays initial>
shows <False>
  <proof>

```

Player1 wins a play if the play has reached a deadlock where it's player0's turn

```
definition player1_wins_immediately :: <'s play  $\Rightarrow$  bool> where
  <player1_wins_immediately play  $\equiv$  player0_position (hd play)  $\wedge$  ( $\nexists$  p' . (hd play)  $\mapsto^{\heartsuit}$  p')>
```

```
definition player0_winning_strategy :: <'s strategy  $\Rightarrow$  's  $\Rightarrow$  bool> where
  <player0_winning_strategy f initial  $\equiv$  ( $\forall$  play  $\in$  plays_for_0strategy f initial.
     $\neg$  player1_wins_immediately play)>
```

```
definition player0_wins :: <'s  $\Rightarrow$  bool> where
  <player0_wins s  $\equiv$  ( $\exists$  f . player0_winning_strategy f s  $\wedge$  sound_0strategy f s)>
```

```
lemma stuck_player0_win:
  assumes <player1_position initial> <( $\nexists$  p' . initial  $\mapsto^{\heartsuit}$  p')>
  shows <player0_wins initial>
  <proof>
```

```
definition player0_wins_immediately :: <'s play  $\Rightarrow$  bool> where
  <player0_wins_immediately play  $\equiv$  player1_position (hd play)  $\wedge$  ( $\nexists$  p' . (hd play)  $\mapsto^{\heartsuit}$  p')>
```

```
end
end
```

## 3 Notions of Equivalence

### 3.1 Strong Simulation and Bisimulation

```
theory Strong_Relations
  imports Transition_Systems
begin
```

```
context lts
begin
```

```
definition simulation ::
  <('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  bool>
where
  <simulation R  $\equiv$   $\forall$  p q. R p q  $\longrightarrow$ 
    ( $\forall$  p' a. p  $\mapsto^a$  p'  $\longrightarrow$ 
      ( $\exists$  q'. R p' q'  $\wedge$  (q  $\mapsto^a$  q')))>
```

```
definition bisimulation ::
  <('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  bool>
where
  <bisimulation R  $\equiv$   $\forall$  p q. R p q  $\longrightarrow$ 
    ( $\forall$  p' a. p  $\mapsto^a$  p'  $\longrightarrow$ 
      ( $\exists$  q'. R p' q'  $\wedge$  (q  $\mapsto^a$  q'))))  $\wedge$ 
    ( $\forall$  q' a. q  $\mapsto^a$  q'  $\longrightarrow$ 
      ( $\exists$  p'. R p' q'  $\wedge$  (p  $\mapsto^a$  p')))>
```

```
lemma bisim_ruleformat:
  assumes <bisimulation R>
  and <R p q>
  shows
    <p  $\mapsto^a$  p'  $\implies$  ( $\exists$  q'. R p' q'  $\wedge$  (q  $\mapsto^a$  q'))>
    <q  $\mapsto^a$  q'  $\implies$  ( $\exists$  p'. R p' q'  $\wedge$  (p  $\mapsto^a$  p'))>
  <proof>
```

```
end — context lts
```

```
end
```

## 3.2 Weak Simulation

```
theory Weak_Relations
```

```
imports
```

```
  Weak_Transition_Systems
```

```
  Strong_Relations
```

```
begin
```

```
context lts_tau
```

```
begin
```

```
definition weak_simulation ::
```

```
  <'s ⇒ 's ⇒ bool> ⇒ bool>
```

```
where
```

```
  <weak_simulation R ≡ ∀ p q. R p q ⟶  
    (∀ p' a. p ⟶a p' ⟶ (∃ q'. R p' q'  
      ∧ (q ⇒~a q')))>
```

Note: Isabelle won't finish the proofs needed for the introduction of the following coinductive predicate if it unfolds the abbreviation of  $\Rightarrow^{\sim}$ . Therefore we use  $\Rightarrow^{\sim\sim}$  as a barrier. There is no mathematical purpose in this.

```
definition weak_step_tau2 :: <'s ⇒ 'a ⇒ 's ⇒ bool>
```

```
  ("_ ⇒~ _ _" [70, 70, 70] 80)
```

```
where [simp]:
```

```
  <(p ⇒~ a q) ≡ p ⇒~a q>
```

```
coinductive greatest_weak_simulation ::
```

```
  <'s ⇒ 's ⇒ bool>
```

```
where
```

```
  <(∀ p' a. p ⟶a p' ⟶ (∃ q'. greatest_weak_simulation p' q' ∧ (q ⇒~ a q'))  
    ⟹ greatest_weak_simulation p q>
```

```
lemma weak_sim_ruleformat:
```

```
assumes <weak_simulation R>
```

```
  and <R p q>
```

```
shows
```

```
  <p ⟶a p' ⟹ ¬tau a ⟹ (∃ q'. R p' q' ∧ (q ⇒~a q'))>
```

```
  <p ⟶a p' ⟹ tau a ⟹ (∃ q'. R p' q' ∧ (q ⟶* tau q'))>
```

```
  <proof>
```

```
abbreviation weakly_simulated_by :: <'s ⇒ 's ⇒ bool> ("_ ⊆ws _" [60, 60] 65)
```

```
  where <weakly_simulated_by p q ≡ ∃ R . weak_simulation R ∧ R p q>
```

```
lemma weaksim_greatest:
```

```
  shows <weak_simulation (λ p q . p ⊆ws q)>
```

```
  <proof>
```

```
lemma gws_is_weak_simulation:
```

```
  shows <weak_simulation greatest_weak_simulation>
```

```
  <proof>
```

```

lemma weakly_sim_by_implies_gws:
  assumes <p  $\sqsubseteq_{ws}$  q>
  shows <greatest_weak_simulation p q>
  <proof>

lemma gws_eq_weakly_sim_by:
  shows <p  $\sqsubseteq_{ws}$  q = greatest_weak_simulation p q>
  <proof>

lemma steps_retain_weak_sim:
  assumes
    <weak_simulation R>
    <R p q>
    <p  $\mapsto^* A$  p'>
    < $\bigwedge a . \tau a \implies A a$ >
  shows < $\exists q' . R p' q' \wedge q \mapsto^* A q'$ >
  <proof>

lemma weak_sim_weak_premise:
  <weak_simulation R =
    ( $\forall p q . R p q \implies
      (\forall p' a . p \Rightarrow^a p' \implies (\exists q' . R p' q' \wedge q \Rightarrow^a q'))$ )>
  <proof>

lemma weak_sim_enabled_subs:
  assumes
    <p  $\sqsubseteq_{ws}$  q>
    <weak_enabled p a>
    < $\neg \tau a$ >
  shows <weak_enabled q a>
  <proof>

lemma weak_sim_union_cl:
  assumes
    <weak_simulation RA>
    <weak_simulation RB>
  shows
    <weak_simulation ( $\lambda p q . RA p q \vee RB p q$ )>
  <proof>

lemma weak_sim_remove_dead_state:
  assumes
    <weak_simulation R>
    < $\bigwedge a p . \neg d \mapsto^a p \wedge \neg p \mapsto^a d$ >
  shows
    <weak_simulation ( $\lambda p q . R p q \wedge q \neq d$ )>
  <proof>

lemma weak_sim_tau_step:
  <weak_simulation ( $\lambda p1 q1 . q1 \mapsto^* \tau p1$ )>
  <proof>

lemma weak_sim_trans_constructive:
  fixes R1 R2
  defines
    <R  $\equiv \lambda p q . \exists pq . (R1 p pq \wedge R2 pq q) \vee (R2 p pq \wedge R1 pq q)$ >
  assumes

```



```

R1_def: <weak_simulation R1> <R1 p pq> and
R2_def: <weak_simulation R2> <R2 pq q>
shows
  <R p q> <weak_simulation R>
<proof>

lemma weak_sim_trans:
  assumes
    <p  $\sqsubseteq_{ws}$  pq>
    <pq  $\sqsubseteq_{ws}$  q>
  shows
    <p  $\sqsubseteq_{ws}$  q>
  <proof>

lemma weak_sim_word_impl:
  fixes
    p q p' A
  assumes
    <weak_simulation R> <R p q> <p  $\Rightarrow$  $ A p'>
  shows
    < $\exists q'. R p' q' \wedge q \Rightarrow$  $ A q'>
  <proof>

lemma weak_sim_word_impl_contra:
  assumes
    < $\forall p q . R p q \longrightarrow$ 
      ( $\forall p' A . p \Rightarrow$  $ A p'  $\longrightarrow$  ( $\exists q' . R p' q' \wedge q \Rightarrow$  $ A q'))>
  shows
    <weak_simulation R>
  <proof>

lemma weak_sim_word:
  <weak_simulation R =
    ( $\forall p q . R p q \longrightarrow$ 
      ( $\forall p' A . p \Rightarrow$  $ A p'  $\longrightarrow$  ( $\exists q' . R p' q' \wedge q \Rightarrow$  $ A q')))>
  <proof>

```

### 3.3 Weak Bisimulation

```

definition weak_bisimulation ::
  <('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  bool>
where
  <weak_bisimulation R  $\equiv$   $\forall p q . R p q \longrightarrow$ 
    ( $\forall p' a . p \longmapsto a p' \longrightarrow$  ( $\exists q' . R p' q' \wedge$ 
      ( $q \Rightarrow \sim a q'$ )))  $\wedge$ 
    ( $\forall q' a . q \longmapsto a q' \longrightarrow$  ( $\exists p' . R p' q' \wedge$ 
      ( $p \Rightarrow \sim a p'$ )))>

lemma weak_bisim_ruleformat:
  assumes <weak_bisimulation R>
  and <R p q>
  shows
    <p  $\longmapsto a p' \implies \neg \tau a \implies$  ( $\exists q' . R p' q' \wedge$  ( $q \Rightarrow a q'$ ))>
    <p  $\longmapsto a p' \implies \tau a \implies$  ( $\exists q' . R p' q' \wedge$  ( $q \longmapsto^* \tau q'$ ))>
    <q  $\longmapsto a q' \implies \neg \tau a \implies$  ( $\exists p' . R p' q' \wedge$  ( $p \Rightarrow a p'$ ))>
    <q  $\longmapsto a q' \implies \tau a \implies$  ( $\exists p' . R p' q' \wedge$  ( $p \longmapsto^* \tau p'$ ))>
  <proof>

```

```

definition tau_weak_bisimulation ::
  <('s ⇒ 's ⇒ bool) ⇒ bool>
where
  <tau_weak_bisimulation R ≡ ∀ p q. R p q ⟶
    (∀ p' a. p ⟶a p' ⟶
      (∃ q'. R p' q' ∧ (q ⇒a q')))) ∧
    (∀ q' a. q ⟶a q' ⟶
      (∃ p'. R p' q' ∧ (p ⇒a p')))>

lemma weak_bisim_implies_tau_weak_bisim:
  assumes
    <tau_weak_bisimulation R>
  shows
    <weak_bisimulation R>
  <proof>

lemma weak_bisim_invert:
  assumes
    <weak_bisimulation R>
  shows
    <weak_bisimulation (λ p q. R q p)>
  <proof>

lemma bisim_weak_bisim:
  assumes <bisimulation R>
  shows <weak_bisimulation R>
  <proof>

lemma weak_bisim_weak_sim:
  shows <weak_bisimulation R = (weak_simulation R ∧ weak_simulation (λ p q . R q p))>
  <proof>

lemma steps_retain_weak_bisim:
  assumes
    <weak_bisimulation R>
    <R p q>
    <p ⟶*A p'>
    <∧ a . tau a ⟹ A a>
  shows <∃ q'. R p' q' ∧ q ⟶*A q'>
  <proof>

lemma weak_bisim_union:
  assumes
    <weak_bisimulation R1>
    <weak_bisimulation R2>
  shows
    <weak_bisimulation (λ p q . R1 p q ∨ R2 p q)>
  <proof>

lemma weak_bisim_taufree_strong:
  assumes
    <weak_bisimulation R>
    <∧ p q a. p ⟶ a q ⟹ ¬ tau a>
  shows
    <bisimulation R>
  <proof>

```

### 3.4 Trace Inclusion

**definition** trace\_inclusion ::  
 <('s ⇒ 's ⇒ bool) ⇒ bool>

**where**

<trace\_inclusion R ≡ ∀ p q p' A . (∀ a ∈ set(A). a ≠ τ)  
 ∧ R p q ∧ p ⇒\$ A p' → (∃ q'. q ⇒\$ A q')>

**abbreviation** weakly\_trace\_included\_by :: <'s ⇒ 's ⇒ bool> ("\_ ⊆T \_" [60, 60] 65)  
**where** <weakly\_trace\_included\_by p q ≡ ∃ R . trace\_inclusion R ∧ R p q>

**lemma** weak\_trace\_inclusion\_greatest:

**shows** <trace\_inclusion (λ p q . p ⊆T q)>  
 <proof>

### 3.5 Delay Simulation

**definition** delay\_simulation ::  
 <('s ⇒ 's ⇒ bool) ⇒ bool>

**where**

<delay\_simulation R ≡ ∀ p q. R p q →  
 (∀ p' a. p ↦<sup>a</sup> p' →  
 (tau a → R p' q) ∧  
 (¬tau a → (∃ q'. R p' q' ∧ (q ⇒<sup>a</sup> q'))))>

**lemma** delay\_simulation\_implies\_weak\_simulation:

**assumes**  
 <delay\_simulation R>  
**shows**  
 <weak\_simulation R>  
 <proof>

### 3.6 Coupled Equivalences

**abbreviation** coupling ::

<('s ⇒ 's ⇒ bool) ⇒ bool>  
**where** <coupling R ≡ ∀ p q . R p q → (∃ q'. q ↦\*tau q' ∧ R q' p)>

**lemma** coupling\_tau\_max\_symm:

**assumes**  
 <R p q → (∃ q'. q ↦\*tau q' ∧ R q' p)>  
 <tau\_max q>  
 <R p q>  
**shows**  
 <R q p>  
 <proof>

**corollary** coupling\_stability\_symm:

**assumes**  
 <R p q → (∃ q'. q ↦\*tau q' ∧ R q' p)>  
 <stable\_state q>  
 <R p q>  
**shows**  
 <R q p>  
 <proof>

**end** — context lts\_tau

**end**

## 4 Contrsimulation

```
theory Contrsimulation
imports
  Weak_Relations
begin
```

```
context lts_tau
begin
```

### 4.1 Definition of Contrsimulation

```
definition contrsimulation ::
  <'s ⇒ 's ⇒ bool> ⇒ bool>
where
  <contrsimulation R ≡ ∀ p q p' A . (∀ a ∈ set(A). a ≠ τ) ∧ R p q ∧ (p ⇒$ A p') →
    (∃ q'. (q ⇒$ A q') ∧ R q' p')>
```

```
lemma contrasim_simpler_def:
  shows <contrsimulation R =
    (∀ p q p' A . R p q ∧ (p ⇒$ A p') → (∃ q'. (q ⇒$ A q') ∧ R q' p'))>
  <proof>
```

```
abbreviation contrasimulated_by :: <'s ⇒ 's ⇒ bool> ("_ ⊆c _" [60, 60] 65)
  where <contrasimulated_by p q ≡ ∃ R . contrsimulation R ∧ R p q>
```

```
lemma contrasim_preorder_is_contrasim:
  shows <contrsimulation (λ p q . p ⊆c q)>
  <proof>
```

```
lemma contrasim_preorder_is_greatest:
  assumes <contrsimulation R>
  shows <∧ p q. R p q ⇒ p ⊆c q>
  <proof>
```

```
lemma contrasim_tau_step:
  <contrsimulation (λ p1 q1 . q1 ⟶* tau p1)>
  <proof>
```

```
lemma contrasim_trans_constructive:
  fixes R1 R2
  defines
    <R ≡ λ p q . ∃ pq . (R1 p pq ∧ R2 pq q) ∨ (R2 p pq ∧ R1 pq q)>
  assumes
    R1_def: <contrsimulation R1> <R1 p pq> and
    R2_def: <contrsimulation R2> <R2 pq q>
  shows
    <R p q> <contrsimulation R>
  <proof>
```

```
lemma contrasim_trans:
  assumes
    <p ⊆c pq>
    <pq ⊆c q>
  shows
    <p ⊆c q>
  <proof>
```

```

lemma contrasim_refl:
  shows
    <p  $\sqsubseteq_c$  p>
  <proof>

lemma contrasimilarity_equiv:
  defines <contrasimilarity  $\equiv \lambda p q. p \sqsubseteq_c q \wedge q \sqsubseteq_c p$ >
  shows <equivp contrasimilarity>
  <proof>

lemma contrasim_implies_trace_incl:
  assumes <contrasimulation R>
  shows <trace_inclusion R>
  <proof>

lemma contrasim_coupled:
  assumes
    <contrasimulation R>
    <R p q>
  shows
    < $\exists q'. q \mapsto^* \text{tau } q' \wedge R q' p$ >
  <proof>

lemma contrasim_taufree_symm:
  assumes
    <contrasimulation R>
    <R p q>
    <stable_state q>
  shows
    <R q p>
  <proof>

lemma symm_contrasim_is_weak_bisim:
  assumes
    <contrasimulation R>
    < $\bigwedge p q. R p q \implies R q p$ >
  shows
    <weak_bisimulation R>
  <proof>

lemma contrasim_weakest_bisim:
  assumes
    <contrasimulation R>
    < $\bigwedge p q a. p \mapsto a q \implies \neg \text{tau } a$ >
  shows
    <bisimulation R>
  <proof>

lemma symm_weak_sim_is_contrasim:
  assumes
    <weak_simulation R>
    < $\bigwedge p q. R p q \implies R q p$ >
  shows
    <contrasimulation R>
  <proof>

```

## 4.2 Intermediate Relation Mimicking Contrasm

```

definition mimicking :: "('s  $\Rightarrow$  's set  $\Rightarrow$  bool)  $\Rightarrow$  's  $\Rightarrow$  's set  $\Rightarrow$  bool" where
<mimicking R p' Q'  $\equiv$   $\exists$ p Q A.
  R p Q  $\wedge$  p  $\Rightarrow$  $A p'  $\wedge$ 
  ( $\forall$ a  $\in$  set A. a  $\neq$   $\tau$ )  $\wedge$ 
  Q' = (dsuccs_seq_rec (rev A) Q)>

```

```

definition set_lifted :: "('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  's  $\Rightarrow$  's set  $\Rightarrow$  bool" where
<set_lifted R p Q  $\equiv$   $\exists$ q. R p q  $\wedge$  Q = {q}>

```

```

lemma R_is_in_mimicking_of_R :
  assumes <R p Q>
  shows <mimicking R p Q>
  <proof>

```

```

lemma mimicking_of_C_guarantees_tau_succ:
  assumes
    <contrasimulation C>
    <mimicking (set_lifted C) p Q>
    <p  $\Rightarrow$   $\tau$  p'>
  shows < $\exists$ q'. q'  $\in$  (weak_tau_succs Q)  $\wedge$  mimicking (set_lifted C) q' {p'}>
  <proof>

```

```

lemma mimicking_of_C_guarantees_action_succ:
  assumes
    <contrasimulation C>
    <mimicking (set_lifted C) p Q>
    <p  $\Rightarrow$  a p'>
    <a  $\neq$   $\tau$ >
  shows <mimicking (set_lifted C) p' (dsuccs a Q)>
  <proof>

```

## 4.3 Over-Approximating Contrasmulation by a Single-Step Version

```

definition contrasim_step ::
  <('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  bool>
where
  <contrasim_step R  $\equiv$   $\forall$  p q p' a .
    R p q  $\wedge$  (p  $\Rightarrow$  a p')  $\longrightarrow$ 
    ( $\exists$  q'. (q  $\Rightarrow$  a q')
       $\wedge$  R q' p')>

```

```

lemma contrasim_step_weaker_than_seq:
  assumes
    <contrasimulation R>
  shows
    <contrasim_step R>
  <proof>

```

```

lemma contrasim_step_seq_coincide_for_sims:
  assumes
    <contrasim_step R>
    <weak_simulation R>
  shows
    <contrasimulation R>
  <proof>

```

```
end
end
```

## 5 Coupled Simulation

```
theory Coupled_Simulation
  imports Contrsimulation
begin
```

```
context lts_tau
begin
```

### 5.1 Van Glabbeek's Coupled Simulation

We mainly use van Glabbeek's coupled simulation from his 2017 CSP paper [7]. Later on, we will compare it to other definitions of coupled (delay/weak) simulations.

```
definition coupled_simulation ::
  <'s ⇒ 's ⇒ bool> ⇒ bool>
where
  <coupled_simulation R ≡ ∀ p q .
    R p q ⟶
      (∀ p' a. p ⟶a p' ⟶
        (∃ q'. R p' q' ∧ q ⇒~a q')) ∧
        (∃ q'. q ⟶*tau q' ∧ R q' p)>
```

```
abbreviation coupled_simulated_by :: <'s ⇒ 's ⇒ bool> ("_ ⊑cs _" [60, 60] 65)
  where <coupled_simulated_by p q ≡ ∃ R . coupled_simulation R ∧ R p q>
```

```
abbreviation coupled_similar :: <'s ⇒ 's ⇒ bool> ("_ ≡cs _" [60, 60] 65)
  where <coupled_similar p q ≡ p ⊑cs q ∧ q ⊑cs p>
```

We call  $\sqsubseteq_{cs}$  "coupled simulation preorder" and  $\equiv_{cs}$  coupled similarity.

### 5.2 Position between Weak Simulation and Weak Bisimulation

Coupled simulations are special weak simulations, and symmetric weak bisimulations also are coupled simulations.

```
lemma coupled_simulation_weak_simulation:
  <coupled_simulation R =
    (weak_simulation R ∧ (∀ p q . R p q ⟶ (∃ q'. q ⟶*tau q' ∧ R q' p)))>
  <proof>
```

```
corollary coupled_simulation_implies_weak_simulation:
  assumes <coupled_simulation R>
  shows <weak_simulation R>
  <proof>
```

```
corollary coupled_sim_enabled_subs:
  assumes
    <p ⊑cs q>
    <weak_enabled p a>
    <¬ tau a>
  shows <weak_enabled q a>
  <proof>
```

**lemma** coupled\_simulation\_implies\_coupling:

```

assumes
  <coupled_simulation R>
  <R p q>
shows
  < $\exists q'. q \mapsto^* \tau q' \wedge R q' p$ >
<proof>

```

**lemma** weak\_bisim\_implies\_coupled\_sim\_gla17:

```

assumes
  wbisim: <weak_bisimulation R> and
  symmetry: < $\bigwedge p q . R p q \implies R q p$ >
  — symmetry is needed here, which is alright because bisimilarity is symmetric.
shows <coupled_simulation R>
<proof>

```

### 5.3 Coupled Simulation and Silent Steps

Coupled simulation shares important patterns with weak simulation when it comes to the treatment of silent steps.

**lemma** coupled\_sim\_step\_gla17:

```

<coupled_simulation ( $\lambda p1 q1 . q1 \mapsto^* \tau p1$ )>
<proof>

```

**corollary** coupled\_sim\_step:

```

assumes
  <p  $\mapsto^* \tau q$ >
shows
  <q  $\sqsubseteq_{cs} p$ >
<proof>

```

A direct implication of this is that states on a tau loop are coupled similar.

**corollary** strongly\_tau\_connected\_coupled\_similar:

```

assumes
  <p  $\mapsto^* \tau q$ >
  <q  $\mapsto^* \tau p$ >
shows <p  $\equiv_{cs} q$ >
<proof>

```

**lemma** silent\_steps\_retain\_coupled\_simulation:

```

assumes
  <coupled_simulation R>
  <R p q>
  <p  $\mapsto^* A p'$ >
  <A = tau>
shows < $\exists q' . q \mapsto^* A q' \wedge R p' q'$ >
<proof>

```

**lemma** coupled\_simulation\_weak\_premise:

```

<coupled_simulation R =
  ( $\forall p q . R p q \implies$ 
    ( $\forall p' a . p \Rightarrow^a p' \implies$ 
      ( $\exists q' . R p' q' \wedge q \Rightarrow^a q'$ ))  $\wedge$ 
      ( $\exists q' . q \mapsto^* \tau q' \wedge R q' p$ ))>
<proof>

```



## 5.4 Closure, Preorder and Symmetry Properties

The coupled simulation preorder  $\sqsubseteq_{cs}$  is a preoder and symmetric at the stable states.

**lemma** coupled\_sim\_union:

```

assumes
  <coupled_simulation R1>
  <coupled_simulation R2>
shows
  <coupled_simulation ( $\lambda p q . R1 p q \vee R2 p q$ )>
<proof>

```

**lemma** coupled\_sim\_refl:

```

<p  $\sqsubseteq_{cs}$  p>
<proof>

```

**lemma** coupled\_sim\_trans:

```

assumes
  <p  $\sqsubseteq_{cs}$  pq>
  <pq  $\sqsubseteq_{cs}$  q>
shows
  <p  $\sqsubseteq_{cs}$  q>
<proof>

```

**interpretation** preorder < $\lambda p q . p \sqsubseteq_{cs} q$ > < $\lambda p q . p \sqsubseteq_{cs} q \wedge \neg(q \sqsubseteq_{cs} p)$ >  
 <proof>

**lemma** coupled\_similarity\_equivalence:

```

<equivp ( $\lambda p q . p \equiv_{cs} q$ )>
<proof>

```

**lemma** coupled\_sim\_tau\_max\_eq:

```

assumes
  <p  $\sqsubseteq_{cs}$  q>
  <tau_max q>
shows <p  $\equiv_{cs}$  q>
<proof>

```

**corollary** coupled\_sim\_stable\_eq:

```

assumes
  <p  $\sqsubseteq_{cs}$  q>
  <stable_state q>
shows <p  $\equiv_{cs}$  q>
<proof>

```

## 5.5 Coinductive Coupled Simulation Preorder

$\sqsubseteq_{cs}$  can also be characterized coinductively.  $\sqsubseteq_{cs}$  is the greatest coupled simulation.

**coinductive** greatest\_coupled\_simulation :: <'s  $\Rightarrow$  's  $\Rightarrow$  bool>

**where** gcs:

```

  <[[ $\bigwedge a p' . p \mapsto a p' \implies \exists q' . q \Rightarrow^{\sim} a q' \wedge$  greatest_coupled_simulation p' q'];
    $\exists q' . q \mapsto^* \text{tau } q' \wedge$  greatest_coupled_simulation q' p]]
   $\implies$  greatest_coupled_simulation p q>

```

**lemma** gcs\_implies\_gws:

```

assumes <greatest_coupled_simulation p q>
shows <greatest_weak_simulation p q>
<proof>

```

```

lemma gcs_is_coupled_simulation:
  shows <coupled_simulation greatest_coupled_simulation>
  <proof>

lemma coupled_similarity_implies_gcs:
  assumes <p  $\sqsubseteq_{cs}$  q>
  shows <greatest_coupled_simulation p q>
  <proof>

lemma gcs_eq_coupled_sim_by:
  shows <p  $\sqsubseteq_{cs}$  q = greatest_coupled_simulation p q>
  <proof>

lemma coupled_sim_by_is_coupled_sim:
  shows
    <coupled_simulation ( $\lambda$  p q . p  $\sqsubseteq_{cs}$  q)>
  <proof>

lemma coupled_sim_unfold:
  shows <p  $\sqsubseteq_{cs}$  q =
    (( $\forall$  a p'. p  $\xrightarrow{a}$  p'  $\longrightarrow$  ( $\exists$  q'. q  $\Rightarrow^a$  q'  $\wedge$  p'  $\sqsubseteq_{cs}$  q'))  $\wedge$ 
    ( $\exists$  q'. q  $\xrightarrow{*}$  tau q'  $\wedge$  q'  $\sqsubseteq_{cs}$  p))>
  <proof>

```

## 5.6 Coupled Simulation Join

The following lemmas reproduce Proposition 3 from [7] that internal choice acts as a least upper bound within the semi-lattice of CSP terms related by  $\sqsubseteq_{cs}$  taking  $\equiv_{cs}$  as equality.

```

lemma coupled_sim_choice_1:
  assumes
    <p  $\sqsubseteq_{cs}$  q>
    < $\bigwedge$  pq a . pqc  $\xrightarrow{a}$  pq  $\longleftrightarrow$  (a =  $\tau$   $\wedge$  (pq = p  $\vee$  pq = q))>
  shows
    <pqc  $\sqsubseteq_{cs}$  q>
    <q  $\sqsubseteq_{cs}$  pqc>
  <proof>

lemma coupled_sim_choice_2:
  assumes
    <pqc  $\sqsubseteq_{cs}$  q>
    < $\bigwedge$  pq a . pqc  $\xrightarrow{a}$  pq  $\longleftrightarrow$  (a =  $\tau$   $\wedge$  (pq = p  $\vee$  pq = q))>
  shows
    <p  $\sqsubseteq_{cs}$  q>
  <proof>

lemma coupled_sim_choice_join:
  assumes
    < $\bigwedge$  pq a . pqc  $\xrightarrow{a}$  pq  $\longleftrightarrow$  (a =  $\tau$   $\wedge$  (pq = p  $\vee$  pq = q))>
  shows
    <p  $\sqsubseteq_{cs}$  q  $\longleftrightarrow$  pqc  $\equiv_{cs}$  q>
  <proof>

```

## 5.7 Coupled Delay Simulation

$\sqsubseteq_{cs}$  can also be characterized in terms of coupled delay simulations, which are conceptionally simpler than van Glabbeek's coupled simulation definition.

In the greatest coupled simulation,  $\tau$ -challenges can be answered by stuttering.

**lemma** `coupledsim_tau_challenge_trivial:`

```

assumes
  <p  $\sqsubseteq_{cs}$  q>
  <p  $\mapsto^* \tau$  p' >
shows
  <p'  $\sqsubseteq_{cs}$  q>
<proof>

```

**lemma** `coupled_similarity_s_delay_simulation:`

```

<delay_simulation ( $\lambda$  p q. p  $\sqsubseteq_{cs}$  q)>
<proof>

```

**definition** `coupled_delay_simulation ::`

```

<('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  bool>
where
  <coupled_delay_simulation R  $\equiv$ 
    delay_simulation R  $\wedge$  coupling R>

```

**lemma** `coupled_sim_by_eq_coupled_delay_simulation:`

```

<(p  $\sqsubseteq_{cs}$  q) = ( $\exists$ R. R p q  $\wedge$  coupled_delay_simulation R)>
<proof>

```

## 5.8 Relationship to Contrasimulation and Weak Simulation

Coupled simulation is precisely the intersection of contrasimulation and weak simulation.

**lemma** `weak_sim_and_contrasim_implies_coupled_sim:`

```

assumes
  <contrasimulation R>
  <weak_simulation R>
shows
  <coupled_simulation R>
<proof>

```

**lemma** `coupledsim_implies_contrasim:`

```

assumes
  <coupled_simulation R>
shows
  <contrasimulation R>
<proof>

```

**lemma** `coupled_simulation_iff_weak_sim_and_contrasim:`

```

shows <coupled_simulation R  $\longleftrightarrow$  contrasimulation R  $\wedge$  weak_simulation R>
<proof>

```

## 5.9 $\tau$ -Reachability (and Divergence)

Coupled similarity comes close to (weak) bisimilarity in two respects:

- If there are no  $\tau$  transitions, coupled similarity coincides with bisimilarity.
- If there are only finite  $\tau$  reachable portions, then coupled similarity contains a bisimilarity on the  $\tau$ -maximal states. (For this,  $\tau$ -cycles have to be ruled out, which, as we show, is no problem because their removal is transparent to coupled similarity.)

**lemma** `taufree_coupledsim_symm:`

```

assumes
  < $\bigwedge p1\ a\ p2 . (p1 \mapsto_a p2 \implies \neg \text{tau } a)$ >
  <coupled_simulation R>
  <R p q>
shows <R q p>
<proof>

```

```

lemma taufree_coupled_sim_weak_bisim:
assumes
  < $\bigwedge p1\ a\ p2 . (p1 \mapsto_a p2 \implies \neg \text{tau } a)$ >
  <coupled_simulation R>
shows <weak_bisimulation R>
<proof>

```

```

lemma coupled_sim_stable_state_symm:
assumes
  <coupled_simulation R>
  <R p q>
  <stable_state q>
shows
  <R q p>
<proof>

```

In finite systems, coupling is guaranteed to happen through  $\tau$ -maximal states.

```

lemma coupled_sim_max_coupled:
assumes
  <p  $\sqsubseteq_{cs}$  q>
  < $\bigwedge r1\ r2 . r1 \mapsto^* \text{tau } r2 \wedge r2 \mapsto^* \text{tau } r1 \implies r1 = r2$ > — contracted tau cycles
  < $\bigwedge r . \text{finite } \{r' . r \mapsto^* \text{tau } r'\}$ >
shows
  < $\exists q' . q \mapsto^* \text{tau } q' \wedge q' \sqsubseteq_{cs} p \wedge \text{tau\_max } q'$ >
<proof>

```

In the greatest coupled simulation, a-challenges can be answered by a weak move without trailing  $\tau$ -steps. (This property is what bridges the gap between weak and delay simulation for coupled simulation.)

```

lemma coupled_sim_step_challenge_short_answer:
assumes
  <p  $\sqsubseteq_{cs}$  q>
  <p  $\mapsto_a p'$ >
  < $\neg \text{tau } a$ >
shows
  < $\exists q'\ q1 . p' \sqsubseteq_{cs} q' \wedge q \mapsto^* \text{tau } q1 \wedge q1 \mapsto_a q'$ >
<proof>

```

If two states share the same outgoing edges with except for one  $\tau$ -loop, then they cannot be distinguished by coupled similarity.

```

lemma coupled_sim_tau_loop_ignorance:
assumes
  < $\bigwedge a\ p' . p \mapsto_a p' \vee p' = pp \wedge a = \text{tau} \iff pp \mapsto_a p'$ >
shows
  <pp  $\equiv_{cs}$  p>
<proof>

```

## 5.10 On the Connection to Weak Bisimulation

When one only considers steps leading to  $\tau$ -maximal states in a system without infinite  $\tau$ -reachable regions (e.g. a finite system), then  $\equiv_{cs}$  on these steps is a bisimulation.

This lemma yields a neat argument why one can use a signature refinement algorithm to pre-select the tuples which come into question for further checking of coupled simulation by contraposition.

```

lemma coupled_sim_eventual_symmetry:
  assumes
    contracted_cycles:  $\langle \bigwedge r_1 r_2 . r_1 \mapsto^* \tau r_2 \wedge r_2 \mapsto^* \tau r_1 \implies r_1 = r_2 \rangle$  and
    finite_taus:  $\langle \bigwedge r . \text{finite } \{r' . r \mapsto^* \tau r'\} \rangle$  and
    cs:  $\langle p \sqsubseteq_{cs} q \rangle$  and
    step:  $\langle p \Rightarrow^a p' \rangle$  and
    tau_max_p':  $\langle \text{tau\_max } p' \rangle$ 
  shows
     $\langle \exists q' . \text{tau\_max } q' \wedge q \Rightarrow^a q' \wedge p' \equiv_{cs} q' \rangle$ 
  (proof)

```

Even without the assumption that the left-hand-side step  $p \Rightarrow^a p'$  ends in a  $\tau$ -maximal state, a situation resembling bismulation can be set up – with the drawback that it only refers to a  $\tau$ -maximal sibling of  $p'$ .

```

lemma coupled_sim_eventuality_2:
  assumes
    contracted_cycles:  $\langle \bigwedge r_1 r_2 . r_1 \mapsto^* \tau r_2 \wedge r_2 \mapsto^* \tau r_1 \implies r_1 = r_2 \rangle$  and
    finite_taus:  $\langle \bigwedge r . \text{finite } \{r' . r \mapsto^* \tau r'\} \rangle$  and
    cbisim:  $\langle p \equiv_{cs} q \rangle$  and
    step:  $\langle p \Rightarrow^a p' \rangle$ 
  shows
     $\langle \exists p'' q' . \text{tau\_max } p'' \wedge \text{tau\_max } q' \wedge p \Rightarrow^a p'' \wedge q \Rightarrow^a q' \wedge p'' \equiv_{cs} q' \rangle$ 
  (proof)

```

```

lemma coupled_sim_eq_reducible_1:
  assumes
    contracted_cycles:  $\langle \bigwedge r_1 r_2 . r_1 \mapsto^* \tau r_2 \wedge r_2 \mapsto^* \tau r_1 \implies r_1 = r_2 \rangle$  and
    finite_taus:  $\langle \bigwedge r . \text{finite } \{r' . r \mapsto^* \tau r'\} \rangle$  and
    tau_shortcuts:
       $\langle \bigwedge r a r' . r \mapsto^* \tau a \implies \exists r'' . \text{tau\_max } r'' \wedge r \mapsto_{\tau} r'' \wedge r' \sqsubseteq_{cs} r'' \rangle$  and
    sim_vis_p:
       $\langle \bigwedge p' a . \neg \text{tau } a \implies p \Rightarrow^a p' \implies \exists p'' q' . q \Rightarrow^a q' \wedge p' \sqsubseteq_{cs} q' \rangle$  and
    sim_tau_max_p:
       $\langle \bigwedge p' . \text{tau\_max } p' \implies p \mapsto^* \tau p' \implies \exists q' . \text{tau\_max } q' \wedge q \mapsto^* \tau q' \wedge p' \equiv_{cs} q' \rangle$ 
  shows
     $\langle p \sqsubseteq_{cs} q \rangle$ 
  (proof)

```

```

lemma coupled_sim_eq_reducible_2:
  assumes
    cs:  $\langle p \sqsubseteq_{cs} q \rangle$  and
    contracted_cycles:  $\langle \bigwedge r_1 r_2 . r_1 \mapsto^* \tau r_2 \wedge r_2 \mapsto^* \tau r_1 \implies r_1 = r_2 \rangle$  and
    finite_taus:  $\langle \bigwedge r . \text{finite } \{r' . r \mapsto^* \tau r'\} \rangle$ 
  shows
    sim_vis_p:
       $\langle \bigwedge p' a . \neg \text{tau } a \implies p \Rightarrow^a p' \implies \exists q' . q \Rightarrow^a q' \wedge p' \sqsubseteq_{cs} q' \rangle$  and
    sim_tau_max_p:
       $\langle \bigwedge p' . \text{tau\_max } p' \implies p \mapsto^* \tau p' \implies \exists q' . \text{tau\_max } q' \wedge q \mapsto^* \tau q' \wedge p' \equiv_{cs} q' \rangle$ 
  (proof)

```

## 5.11 Reduction Semantics Coupled Simulation

The tradition to describe coupled simulation as special delay/weak simulation is quite common for coupled simulations on reduction semantics as in [11, 6], of which [11] can also be found in the AFP [12]. The notions coincide (for systems just with  $\tau$ -transitions).

```

definition coupled_simulation_gp15 ::
  <('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  bool>
where
  <coupled_simulation_gp15 R  $\equiv$   $\forall$  p q p'. R p q  $\wedge$  (p  $\mapsto^*$  ( $\lambda$ a. True) p')  $\longrightarrow$ 
    ( $\exists$  q'. (q  $\mapsto^*$  ( $\lambda$ a. True) q')  $\wedge$  R p' q')  $\wedge$ 
    ( $\exists$  q'. (q  $\mapsto^*$  ( $\lambda$ a. True) q')  $\wedge$  R q' p')>

```

**lemma** weak\_bisim\_implies\_coupled\_sim\_gp15:

```

assumes
  wbsim: <weak_bisimulation R> and
  symmetry: < $\bigwedge$  p q . R p q  $\implies$  R q p>
shows <coupled_simulation_gp15 R>
<proof>

```

**lemma** coupled\_sim\_gla17\_implies\_gp15:

```

assumes
  <coupled_simulation R>
shows
  <coupled_simulation_gp15 R>
<proof>

```

**lemma** coupled\_sim\_gp15\_implies\_gla17\_on\_tau\_systems:

```

assumes
  <coupled_simulation_gp15 R>
  < $\bigwedge$  a . tau a>
shows
  <coupled_simulation R>
<proof>

```

## 5.12 Coupled Simulation as Two Simulations

Historically, coupled similarity has been defined in terms of *two* weak simulations coupled in some way [14, 10]. We reproduce these (more well-known) formulations and show that they are equivalent to the coupled (delay) simulations we are using.

```

definition coupled_simulation_san12 ::
  <('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  ('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  bool>
where
  <coupled_simulation_san12 R1 R2  $\equiv$ 
    weak_simulation R1  $\wedge$  weak_simulation ( $\lambda$  p q . R2 q p)
   $\wedge$  ( $\forall$  p q . R1 p q  $\longrightarrow$  ( $\exists$  q' . q  $\mapsto^*$  tau q'  $\wedge$  R2 p q'))
   $\wedge$  ( $\forall$  p q . R2 p q  $\longrightarrow$  ( $\exists$  p' . p  $\mapsto^*$  tau p'  $\wedge$  R1 p' q))>

```

**lemma** weak\_bisim\_implies\_coupled\_sim\_san12:

```

assumes <weak_bisimulation R>
shows <coupled_simulation_san12 R R>
<proof>

```

**lemma** coupled\_sim\_gla17\_resembles\_san12:

```

shows
  <coupled_simulation R1 =
    coupled_simulation_san12 R1 ( $\lambda$  p q . R1 q p)>

```

*<proof>*

```
lemma coupled_sim_san12_impl_gla17:
  assumes
    <coupled_simulation_san12 R1 R2>
  shows
    <coupled_simulation (λ p q. R1 p q ∨ R2 q p)>
  <proof>
```

### 5.13 S-coupled Simulation

Originally coupled simulation was introduced as two weak simulations coupled at the stable states. We give the definitions from [9, 10] and a proof connecting this notion to “our” coupled similarity in the absence of divergences following [14].

```
definition coupled_simulation_p92 ::
  <('s ⇒ 's ⇒ bool) ⇒ ('s ⇒ 's ⇒ bool) ⇒ bool>
where
  <coupled_simulation_p92 R1 R2 ≡ ∀ p q .
    (R1 p q →
      ((∀ p' a. p → a p' →
        (∃ q'. R1 p' q' ∧
          (q ⇒ ¬a q')))) ∧
      (stable_state p → R2 p q))) ∧
    (R2 p q →
      ((∀ q' a. q → a q' →
        (∃ p'. R2 p' q' ∧
          (p ⇒ ¬a p')))) ∧
      (stable_state q → R1 p q)))>
```

```
lemma weak_bisim_implies_coupled_sim_p92:
  assumes <weak_bisimulation R>
  shows <coupled_simulation_p92 R R>
  <proof>
```

```
lemma coupled_sim_p92_symm:
  assumes <coupled_simulation_p92 R1 R2>
  shows <coupled_simulation_p92 (λ p q. R2 q p) (λ p q. R1 q p)>
  <proof>
```

```
definition s_coupled_simulation_san12 ::
  <('s ⇒ 's ⇒ bool) ⇒ ('s ⇒ 's ⇒ bool) ⇒ bool>
where
  <s_coupled_simulation_san12 R1 R2 ≡
    weak_simulation R1 ∧ weak_simulation (λ p q . R2 q p)
  ∧ (∀ p q . R1 p q → stable_state p → R2 p q)
  ∧ (∀ p q . R2 p q → stable_state q → R1 p q)>
```

```
abbreviation s_coupled_simulated_by :: <'s ⇒ 's ⇒ bool> ("_ ⊑scs _" [60, 60] 65)
where <s_coupled_simulated_by p q ≡
  ∃ R1 R2 . s_coupled_simulation_san12 R1 R2 ∧ R1 p q>
```

```
abbreviation s_coupled_similar :: <'s ⇒ 's ⇒ bool> ("_ ≡scs _" [60, 60] 65)
where <s_coupled_similar p q ≡
  ∃ R1 R2 . s_coupled_simulation_san12 R1 R2 ∧ R1 p q ∧ R2 p q>
```

```
lemma s_coupled_sim_is_original_coupled:
  <s_coupled_simulation_san12 = coupled_simulation_p92>
```

```

<proof>

corollary weak_bisim_implies_s_coupled_sim:
  assumes <weak_bisimulation R>
  shows <s_coupled_simulation_san12 R R>
  <proof>

corollary s_coupled_sim_symm:
  assumes <s_coupled_simulation_san12 R1 R2>
  shows <s_coupled_simulation_san12 ( $\lambda p q. R2 q p$ ) ( $\lambda p q. R1 q p$ )>
  <proof>

corollary s_coupled_sim_union_cl:
  assumes
    <s_coupled_simulation_san12 RA1 RA2>
    <s_coupled_simulation_san12 RB1 RB2>
  shows
    <s_coupled_simulation_san12 ( $\lambda p q. RA1 p q \vee RB1 p q$ ) ( $\lambda p q. RA2 p q \vee RB2 p q$ )>
  <proof>

corollary s_coupled_sim_symm_union:
  assumes <s_coupled_simulation_san12 R1 R2>
  shows <s_coupled_simulation_san12 ( $\lambda p q. R1 p q \vee R2 q p$ ) ( $\lambda p q. R2 p q \vee R1 q p$ )>
  <proof>

lemma s_coupledsim_stable_eq:
  assumes
    <p  $\sqsubseteq_{scs}$  q>
    <stable_state p>
  shows <p  $\equiv_{scs}$  q>
  <proof>

lemma s_coupledsim_symm:
  assumes
    <p  $\equiv_{scs}$  q>
  shows
    <q  $\equiv_{scs}$  p>
  <proof>

lemma s_coupledsim_eq_parts:
  assumes
    <p  $\equiv_{scs}$  q>
  shows
    <p  $\sqsubseteq_{scs}$  q>
    <q  $\sqsubseteq_{scs}$  p>
  <proof>

lemma divergence_free_coupledsims_coincidence_1:
  defines
    <R1  $\equiv (\lambda p q . p \sqsubseteq_{cs} q \wedge (stable\_state p \longrightarrow stable\_state q))$ > and
    <R2  $\equiv (\lambda p q . q \sqsubseteq_{cs} p \wedge (stable\_state q \longrightarrow stable\_state p))$ >
  assumes
    non_divergent_system: < $\bigwedge p . \neg divergent\_state p$ >
  shows
    <s_coupled_simulation_san12 R1 R2>
  <proof>

lemma divergence_free_coupledsims_coincidence_2:
  defines

```



```

  <R ≡ (λ p q . p ⊆scs q ∨ (∃ q' . q ⟶* tau q' ∧ p ≡scs q'))>
  assumes
    non_divergent_system: <λ p . ¬ divergent_state p>
  shows
    <coupled_simulation R>
  <proof>

```

While this proof follows [14], we needed to deviate from them by also requiring rootedness (shared stability) for the compared states.

```

theorem divergence_free_coupled_sims_coincidence:
  assumes
    non_divergent_system: <λ p . ¬ divergent_state p> and
    stability_rooted: <stable_state p ⟷ stable_state q>
  shows
    <(p ≡cs q) = (p ≡scs q)>
  <proof>

```

end — context lts\_tau

The following example shows that a system might be related by s-coupled-simulation without being connected by coupled-simulation.

```

datatype ex_state = a0 | a1 | a2 | a3 | b0 | b1 | b2

```

```

locale ex_lts = lts_tau trans τ
  for trans :: <ex_state ⇒ nat ⇒ ex_state ⇒ bool> ("_ ⟶_ _" [70, 70, 70] 80) and τ +
  assumes
    sys:
      <trans = (λ p act q .
        1 = act ∧ (p = a0 ∧ q = a1
          ∨ p = a0 ∧ q = a2
          ∨ p = a2 ∧ q = a3
          ∨ p = b0 ∧ q = b1
          ∨ p = b1 ∧ q = b2) ∨
        0 = act ∧ (p = a1 ∧ q = a1))>
      <τ = 0>
  begin

```

```

lemma no_root_coupled_sim:
  fixes R1 R2
  assumes
    coupled:
      <coupled_simulation_san12 R1 R2> and
    root:
      <R1 a0 b0> <R2 a0 b0>
  shows
    False
  <proof>

```

```

lemma root_s_coupled_sim:
  defines
    <R1 ≡ λ a b .
      a = a0 ∧ b = b0 ∨
      a = a1 ∧ b = b1 ∨
      a = a2 ∧ b = b1 ∨
      a = a3 ∧ b = b2>
  and
    <R2 ≡ λ a b .

```

```

    a = a0 ∧ b = b0 ∨
    a = a2 ∧ b = b1 ∨
    a = a3 ∧ b = b2>
shows
  coupled:
    <s_coupled_simulation_san12 R1 R2>
  <proof>

end — ex_lts// example lts

end

```

## 6 Game for Coupled Similarity with Delay Formulation

```

theory CoupledSim_Game_Delay
imports
  Coupled_Simulation
  Simple_Game
begin

```

### 6.1 The Coupled Simulation Preorder Game Using Delay Steps

```

datatype ('s, 'a) cs_game_node =
  AttackerNode 's 's |
  DefenderStepNode 'a 's 's |
  DefenderCouplingNode 's 's

fun (in lts_tau) cs_game_moves ::
  <('s, 'a) cs_game_node ⇒ ('s, 'a) cs_game_node ⇒ bool> where
  simulation_visible_challenge:
    <cs_game_moves (AttackerNode p q) (DefenderStepNode a p1 q0) =
      (¬tau a ∧ p ⟶ a p1 ∧ q = q0)> |
  simulation_internal_attacker_move:
    <cs_game_moves (AttackerNode p q) (AttackerNode p1 q0) =
      (∃a. tau a ∧ p ⟶ a p1 ∧ q = q0)> |
  simulation_answer:
    <cs_game_moves (DefenderStepNode a p1 q0) (AttackerNode p11 q1) =
      (q0 ⟶ a q1 ∧ p1 = p11)> |
  coupling_challenge:
    <cs_game_moves (AttackerNode p q) (DefenderCouplingNode p0 q0) =
      (p = p0 ∧ q = q0)> |
  coupling_answer:
    <cs_game_moves (DefenderCouplingNode p0 q0) (AttackerNode q1 p00) =
      (p0 = p00 ∧ q0 ⟶* tau q1)> |
  cs_game_moves_no_step:
    <cs_game_moves _ _ = False>

fun cs_game_defender_node :: <('s, 'a) cs_game_node ⇒ bool> where
  <cs_game_defender_node (AttackerNode _ _) = False> |
  <cs_game_defender_node (DefenderStepNode _ _ _) = True> |
  <cs_game_defender_node (DefenderCouplingNode _ _) = True>

locale cs_game =
  lts_tau trans τ +
  simple_game cs_game_moves cs_game_defender_node
for
  trans :: <'s ⇒ 'a ⇒ 's ⇒ bool> ("_ ⟶_ _" [70, 70, 70] 80) and

```

```

 $\tau$  :: <'a>
begin

```

## 6.2 Coupled Simulation Implies Winning Strategy

```

fun strategy_from_coupleddsim :: <('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  ('s, 'a) cs_game_node strategy> where
  <strategy_from_coupleddsim R ((DefenderStepNode a p1 q0)#play) =
    (AttackerNode p1 (SOME q1 . R p1 q1  $\wedge$  q0  $\Rightarrow$ a q1))> |
  <strategy_from_coupleddsim R ((DefenderCouplingNode p0 q0)#play) =
    (AttackerNode (SOME q1 . R q1 p0  $\wedge$  q0  $\mapsto^*$  tau q1) p0)> |
  <strategy_from_coupleddsim _ _ = undefined>

```

```

lemma defender_preceded_by_attacker:

```

```

  assumes

```

```

    <n0 # play  $\in$  plays (AttackerNode p0 q0)>
    <cs_game_defender_node n0>

```

```

  shows

```

```

    < $\exists$  p q . hd play = AttackerNode p q  $\wedge$  cs_game_moves (AttackerNode p q) n0>
    <play  $\neq$  []>

```

```

  <proof>

```

```

lemma defender_only_challenged_by_visible_actions:

```

```

  assumes

```

```

    <(DefenderStepNode a p q) # play  $\in$  plays (AttackerNode p0 q0)>

```

```

  shows

```

```

    < $\neg$ tau a>

```

```

  <proof>

```

```

lemma strategy_from_coupleddsim_retains_coupleddsim:

```

```

  assumes

```

```

    <R p0 q0>
    <coupled_delay_simulation R>
    <initial = AttackerNode p0 q0>
    <play  $\in$  plays_for_0strategy (strategy_from_coupleddsim R) initial>

```

```

  shows

```

```

    <hd play = AttackerNode p q  $\implies$  R p q>
    <length play > 1  $\implies$  hd (tl play) = AttackerNode p q  $\implies$  R p q>

```

```

  <proof>

```

```

lemma strategy_from_coupleddsim_sound:

```

```

  assumes

```

```

    <R p0 q0>
    <coupled_delay_simulation R>
    <initial = AttackerNode p0 q0>

```

```

  shows

```

```

    <sound_0strategy (strategy_from_coupleddsim R) initial>

```

```

  <proof>

```

```

lemma coupleddsim_implies_winning_strategy:

```

```

  assumes

```

```

    <R p q>
    <coupled_delay_simulation R>
    <initial = AttackerNode p q>

```

```

  shows

```

```

    <player0_winning_strategy (strategy_from_coupleddsim R) initial>

```

```

  <proof>

```

## 6.3 Winning Strategy Induces Coupled Simulation

```

lemma winning_strategy_implies_coupleddsim:
  assumes
    <player0_winning_strategy f initial>
    <sound_0strategy f initial>
  defines
    <R ==  $\lambda$  p q . ( $\exists$  play  $\in$  plays_for_0strategy f initial. hd play = AttackerNode p q)>
  shows
    <coupled_delay_simulation R>
  <proof>

theorem winning_strategy_iff_coupleddsim:
  assumes
    <initial = AttackerNode p q>
  shows
    <( $\exists$  f . player0_winning_strategy f initial  $\wedge$  sound_0strategy f initial)
      = p  $\sqsubseteq_{cs}$  q>
  <proof>

end
end

```

## 7 Fixed Point Algorithm for Coupled Similarity

### 7.1 The Algorithm

```

theory Coupleddsim_Fixpoint_Algo_Delay
imports
  Coupled_Simulation
  "HOL-Library.While_Combinator"
  "HOL-Library.Finite_Lattice"
begin

context lts_tau
begin

definition fp_step :: <'s rel  $\Rightarrow$  's rel>
  <'s rel  $\Rightarrow$  's rel>
where
  <fp_step R1  $\equiv$  { (p,q) $\in$ R1.
    ( $\forall$  p' a. p  $\mapsto$ a p'  $\rightarrow$ 
      (tau a  $\rightarrow$  (p',q) $\in$ R1)  $\wedge$ 
      ( $\neg$ tau a  $\rightarrow$  ( $\exists$  q'. ((p',q') $\in$ R1)  $\wedge$  (q  $\Rightarrow$ a q'))))  $\wedge$ 
    ( $\exists$  q'. q  $\mapsto$ *tau q'  $\wedge$  ((q',p) $\in$ R1)) }>

definition fp_compute_cs :: <'s rel>
where <fp_compute_cs  $\equiv$  while ( $\lambda$ R. fp_step R  $\neq$  R) fp_step top>

```

### 7.2 Correctness

```

lemma mono_fp_step:
  <mono fp_step>
  <proof>

lemma fp_fp_step:
  assumes

```

```

    <R = fp_step R>
  shows
    <coupled_delay_simulation (λ p q. (p, q) ∈ R)>
    <proof>

lemma gfp_fp_step_subset_gcs:
  shows <(gfp fp_step) ⊆ { (p,q) . greatest_coupled_simulation p q }>
  <proof>

lemma fp_fp_step_gcs:
  assumes
    <R = { (p,q) . greatest_coupled_simulation p q }>
  shows
    <fp_step R = R>
  <proof>

lemma gfp_fp_step_gcs: <gfp fp_step = { (p,q) . greatest_coupled_simulation p q }>
  <proof>

end

context lts_tau_finite
begin
lemma gfp_fp_step_while:
  shows
    <gfp fp_step = fp_compute_cs>
  <proof>

theorem coupled_sim_fp_step_while:
  shows <fp_compute_cs = { (p,q) . greatest_coupled_simulation p q }>
  <proof>

end

end

```

## 8 The Contrsimulation Preorder Word Game

```

theory Contrsim_Word_Game
imports
  Simple_Game
  Contrsimulation
begin

datatype ('s, 'a) c_word_game_node =
  AttackerNode 's 's |
  DefenderNode "'a list" 's 's

fun (in lts_tau) c_word_game_moves ::
  <('s, 'a) c_word_game_node ⇒ ('s, 'a) c_word_game_node ⇒ bool> where

  simulation_challenge:
    <c_word_game_moves (AttackerNode p q) (DefenderNode A p1 q0) =
      (p ⇒ $A p1 ∧ q = q0 ∧ (∀ a ∈ set A. a ≠ τ))> |

  simulation_answer:

```

```

    <c_word_game_moves (DefenderNode A p1 q0) (AttackerNode q1 p10) =
      (q0  $\Rightarrow$  $A q1  $\wedge$  p1 = p10) > |

c_word_game_moves_no_step:
  <c_word_game_moves _ _ = False >

fun c_word_game_defender_node :: <('s, 'a) c_word_game_node  $\Rightarrow$  bool > where
  <c_word_game_defender_node (AttackerNode _ _) = False > |
  <c_word_game_defender_node (DefenderNode _ _ _) = True >

```

## 8.1 Contrsimulation Implies Winning Strategy in Word Game (Completeness)

```

locale c_word_game =
  lts_tau trans  $\tau$  +
  simple_game c_word_game_moves c_word_game_defender_node
for
  trans :: <'s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool > and
   $\tau$  :: <'a > and
  initial :: <('s, 'a) c_word_game_node >
begin

fun strategy_from_contrasim :: <('s  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  ('s, 'a) c_word_game_node strategy > where
  <strategy_from_contrasim R ((DefenderNode A p1 q0)#play) =
    (AttackerNode (SOME q1 . R q1 p1  $\wedge$  q0  $\Rightarrow$  $A q1) p1) > |
  <strategy_from_contrasim _ _ = undefined >

lemma cwg_atknodes_precede_defnodes_in_plays:
  assumes
    <c_word_game_defender_node n0 >
    <n0 = DefenderNode A p' q' >
    <(n0#play)  $\in$  plays initial >
    <initial = AttackerNode p0 q0 >
  shows < $\exists$ p. (hd play) = AttackerNode p q  $\wedge$  c_word_game_moves (hd play) n0 >
  <proof >

lemma cwg_second_play_elem_in_play_set :
  assumes
    <(n0#play)  $\in$  plays initial >
    <initial = AttackerNode p0 q0 >
    <n0 = DefenderNode A p q >
  shows <hd play  $\in$  set (n0 # play) >
  <proof >

lemma cwg_contrasim_contains_all_strat_consistent_atknodes:
  assumes
    <contrasimulation R >
    <R p0 q0 >
    <initial = AttackerNode p0 q0 >
    <play  $\in$  plays_for_Ostrategy (strategy_from_contrasim R) initial >
  shows <((AttackerNode p q)  $\in$  set play)  $\implies$  R p q >
  <proof >

lemma contrasim_word_game_complete:
  assumes
    <contrasimulation R >
    <R p q >

```

```

    <initial = AttackerNode p q>
  shows <player0_winning_strategy (strategy_from_contrasim R) initial>
    <proof>

```

## 8.2 Winning Strategy Implies Contrsimulation in Word Game (Soundness)

```

lemma cwg_strategy_from_contrasim_sound:

```

```

  assumes
    <R p0 q0>
    <contrasimulation R>
    <initial = AttackerNode p0 q0>
  shows
    <sound_0strategy (strategy_from_contrasim R) initial>
    <proof>

```

```

lemma contrasim_word_game_sound:

```

```

  assumes
    <player0_winning_strategy f initial>
    <sound_0strategy f initial>
  defines
    <R == λ p q . (∃ play ∈ plays_for_0strategy f initial. hd play = AttackerNode p q)>
  shows
    <contrasimulation R> <proof>

```

```

theorem winning_strategy_in_c_word_game_iff_contrasim:

```

```

  assumes
    <initial = AttackerNode p q>
  shows
    <(∃ f . player0_winning_strategy f initial ∧ sound_0strategy f initial)
    = (∃ C. contrasimulation C ∧ C p q)>
    <proof>

```

```

end
end

```

## 9 The Contrsimulation Preorder Set Game

```

theory Contrsim_Set_Game

```

```

imports
  Simple_Game
  Contrsimulation
begin

```

```

datatype ('s, 'a) c_set_game_node =
  AttackerNode 's "'s set" |
  DefenderSimNode 'a 's "'s set" |
  DefenderSwapNode 's "'s set"

```

```

fun (in lts_tau) c_set_game_moves ::
  <('s, 'a) c_set_game_node ⇒ ('s, 'a) c_set_game_node ⇒ bool> where

```

```

  simulation_challenge:
    <c_set_game_moves (AttackerNode p Q) (DefenderSimNode a p1 Q0) =
    (p =>a p1 ∧ Q = Q0 ∧ ¬ tau a)> |

```

```

simulation_answer:
  <c_set_game_moves (DefenderSimNode a p1 Q) (AttackerNode p10 Q1) =
    (p1 = p10  $\wedge$  Q1 = dsuccs a Q)> |

swap_challenge:
  <c_set_game_moves (AttackerNode p Q) (DefenderSwapNode p1 Q0) =
    (p  $\Rightarrow^{\tau}$  p1  $\wedge$  Q = Q0)> |

swap_answer:
  <c_set_game_moves (DefenderSwapNode p1 Q) (AttackerNode q1 P1) =
    (q1  $\in$  weak_tau_succs Q  $\wedge$  P1 = {p1})> |

c_set_game_moves_no_step:
  <c_set_game_moves _ _ = False>

fun c_set_game_defender_node :: <('s, 'a) c_set_game_node  $\Rightarrow$  bool> where
  <c_set_game_defender_node (AttackerNode _ _) = False> |
  <c_set_game_defender_node (DefenderSimNode _ _ _) = True> |
  <c_set_game_defender_node (DefenderSwapNode _ _ _) = True>

```

## 9.1 Contrsimulation Implies Winning Strategy in Set Game (Completeness)

```

locale c_set_game =
  lts_tau trans  $\tau$  +
  simple_game c_set_game_moves c_set_game_defender_node
for
  trans :: <'s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool> and
   $\tau$  :: <'a>
begin

fun strategy_from_mimicking_of_C ::
  <('s  $\Rightarrow$  ('s set)  $\Rightarrow$  bool)  $\Rightarrow$  ('s, 'a) c_set_game_node strategy>
  where

    <strategy_from_mimicking_of_C R ((DefenderSwapNode p1 Q)#play) =
      (AttackerNode (SOME q1 . ( $\exists$ q. (q  $\in$  Q  $\wedge$  q  $\Rightarrow^{\tau}$  q1))  $\wedge$  R q1 {p1}) {p1})> |

    <strategy_from_mimicking_of_C R ((DefenderSimNode a p1 Q)#play) =
      (AttackerNode p1 (SOME Q1 . Q1 = dsuccs a Q  $\wedge$  R p1 Q1))> |

    <strategy_from_mimicking_of_C _ _ = undefined>

lemma csg_atknodes_precede_defnodes_in_plays:
  assumes
    <c_set_game_defender_node n0>
    <(n0#play)  $\in$  plays (AttackerNode p0 Q0)>
  shows < $\exists$ p Q. (hd play) = AttackerNode p Q  $\wedge$  c_set_game_moves (hd play) n0>
  <proof>

lemma csg_second_play_elem_in_play_set:
  assumes
    <(n0#play)  $\in$  plays (AttackerNode p0 Q0)>
    <c_set_game_defender_node n0>
  shows
    <hd play  $\in$  set (n0 # play)>
  <proof>

```



```

lemma csg_only_defnodes_move_to_atknodes:
  assumes
    <c_set_game_moves n0 n1>
    <n1 = AttackerNode p Q>
  shows
    <( $\exists$  Qpred a. n0 = (DefenderSimNode a p Qpred))  $\vee$ 
    ( $\exists$  q Ppred. n0 = (DefenderSwapNode q Ppred)  $\wedge$  Q = {q})>
  <proof>

lemma c_set_game_strategy_retains_mimicking:
  assumes
    <contrasimulation C>
    <C p0 q0>
    <play  $\in$  plays_for_Ostrategy
    (strategy_from_mimicking_of_C (mimicking (set_lifted C))) (AttackerNode p0 {q0})>
  shows
    <n = AttackerNode p Q  $\implies$  n  $\in$  set play  $\implies$  mimicking (set_lifted C) p Q >
  <proof>

lemma contrasim_set_game_complete:
  assumes
    <contrasimulation C>
    <C p0 q0>
  shows
    <player0_winning_strategy (strategy_from_mimicking_of_C
    (mimicking (set_lifted C))) (AttackerNode p0 {q0})>
  <proof>

lemma csg_strategy_from_mimicking_of_C_sound:
  assumes
    <contrasimulation C>
    <C p0 q0>
  shows
    <sound_Ostrategy
    (strategy_from_mimicking_of_C (mimicking (set_lifted C)))
    (AttackerNode p0 {q0})>
  <proof>

```

## 9.2 Winning Strategy Implies Contrasimulation in Set Game (Soundness)

```

lemma csg_move_defsimnode_to_atknode:
  assumes
    <c_set_game_moves (DefenderSimNode a p Q) n0>
  shows
    <n0 = AttackerNode p (dsuccs a Q)>
  <proof>

lemma csg_move_defswapnode_to_atknode:
  assumes
    <c_set_game_moves (DefenderSwapNode p' Q) n0>
  shows
    < $\exists$  q'. n0 = AttackerNode q' {p'}  $\wedge$  q'  $\in$  weak_tau_succs Q>
  <proof>

lemma csg_defsimnode_never_stuck:

```

```

    assumes <n0 = DefenderSimNode a p Q>
    shows < $\exists Q'. c\_set\_game\_moves\ n0\ (AttackerNode\ p\ Q')>$ 
  <proof>

lemma csg_defender_can_simulate_prefix:
  assumes
    <A  $\neq$  []>
    <p  $\Rightarrow$  $A p1>
    < $\forall a \in \text{set } A. a \neq \tau$ >
    <sound_Ostrategy f (AttackerNode p00 {q00})>
    <play  $\in$  plays_for_Ostrategy f (AttackerNode p00 {q00})>
    <hd play = AttackerNode p {q}>
  shows
    < $\exists$  play p0.
      ((DefenderSimNode (last A) p0 (dsuccs_seq_rec (rev (butlast A)) {q}))#play)
         $\in$  plays_for_Ostrategy f (AttackerNode p00 {q00})
         $\wedge$  word_reachable_via_delay A p p0 p1>
  <proof>

lemma contrasim_set_game_sound:
  assumes
    <player0_winning_strategy f (AttackerNode p00 {q00})>
    <sound_Ostrategy f (AttackerNode p00 {q00})>
  defines
    <C ==  $\lambda p\ q. (\exists \text{ play} \in \text{plays\_for\_Ostrategy } f\ (\text{AttackerNode } p00\ \{q00\}) .$ 
      hd play = AttackerNode p {q}  $\wedge$  (hd play = (AttackerNode p00 {q00})
         $\vee$  ( $\exists P. \text{hd } (t1\ \text{play}) = \text{DefenderSwapNode } q\ P))$ >
  shows
    <contrasimulation C>
  <proof>

theorem winning_strategy_in_c_set_game_iff_contrasim:
  shows
    <( $\exists f. \text{player0\_winning\_strategy } f\ (\text{AttackerNode } p0\ \{q0\})$ 
       $\wedge$  sound_Ostrategy f (AttackerNode p0 {q0}))
      = p0  $\sqsubseteq_c$  q0>
  <proof>

end
end

```

## 10 Infinitary Hennessy–Milner Logic

```

theory HM_Logic_Infinitary
  imports
    Weak_Relations
begin

datatype ('a,'x)HML_formula =
  HML_true
| HML_conj <'x set> <'x  $\Rightarrow$  ('a,'x)HML_formula> (<AND _ _>)
| HML_neg <('a,'x)HML_formula> (<~_> [20] 60)
| HML_poss <'a> <('a,'x)HML_formula> (<(_)> [60] 60)

```

— The HML formulation is derived from that by Max Pohlmann [13].

## 10.1 Satisfaction Relation

```

context lts_tau
begin

function satisfies :: <'s ⇒ ('a, 's) HML_formula ⇒ bool>
  (<_ ⊨ _> [50, 50] 50)
  where
    <p ⊨ HML_true> = True>
  | <p ⊨ HML_conj I F> = (∀ i ∈ I. p ⊨ (F i))>
  | <p ⊨ HML_neg φ> = (¬ p ⊨ φ)>
  | <p ⊨ HML_poss α φ> =
    (∃ p'. ((tau α ∧ p ⟶* tau p') ∨ (¬ tau α ∧ p ⟶α p')) ∧ p' ⊨ φ)>
  <proof>

inductive_set HML_wf_rel :: <('s × ('a, 's) HML_formula) rel>
  where
    <φ = F i ∧ i ∈ I ⇒ ((p, φ), (p, HML_conj I F)) ∈ HML_wf_rel>
  | <((p, φ), (p, HML_neg φ)) ∈ HML_wf_rel>
  | <((p, φ), (p', HML_poss α φ)) ∈ HML_wf_rel>

lemma HML_wf_rel_is_wf: <wf HML_wf_rel>
  <proof>

termination satisfies <proof>

inductive_set HML_direct_subformulas :: <(('a, 's) HML_formula) rel>
  where
    <φ = F i ∧ i ∈ I ⇒ (φ, HML_conj I F) ∈ HML_direct_subformulas>
  | <(φ, HML_neg φ) ∈ HML_direct_subformulas>
  | <(φ, HML_poss α φ) ∈ HML_direct_subformulas>

lemma HML_direct_subformulas_wf: <wf HML_direct_subformulas>
  <proof>

definition HML_subformulas where <HML_subformulas ≡ (HML_direct_subformulas)+>

lemma HML_subformulas_wf: <wf HML_subformulas>
  <proof>

lemma conj_only_depends_on_indexset:
  assumes <∀ i ∈ I. f1 i = f2 i>
  shows <(p ⊨ HML_conj I f1) = (p ⊨ HML_conj I f2)>
  <proof>

```

## 10.2 Distinguishing Formulas

```

definition HML_equivalent :: <'s ⇒ 's ⇒ bool>
  where <HML_equivalent p q
    ≡ (∀ φ :: ('a, 's) HML_formula. (p ⊨ φ) ↔ (q ⊨ φ))>

fun distinguishes :: <('a, 's) HML_formula ⇒ 's ⇒ 's ⇒ bool>
  where
    <distinguishes φ p q = (p ⊨ φ ∧ ¬ q ⊨ φ)>

fun distinguishes_from_set :: <('a, 's) HML_formula ⇒ 's ⇒ 's set ⇒ bool>
  where
    <distinguishes_from_set φ p Q = (p ⊨ φ ∧ (∀ q. q ∈ Q ⟶ ¬ q ⊨ φ))>

```

```

lemma distinguishing_formula:
  assumes <math>\neg \text{HML\_equivalent } p \ q</math>
  shows <math>\exists \varphi. p \models \varphi \wedge \neg q \models \varphi</math>
  <math>\langle \text{proof} \rangle</math>

```

```

lemma HML_equivalent_symm:
  assumes <math>\text{HML\_equivalent } p \ q</math>
  shows <math>\text{HML\_equivalent } q \ p</math>
  <math>\langle \text{proof} \rangle</math>

```

### 10.3 Weak-NOR Hennessy–Milner Logic

```

definition HML_weaknor ::
  <math>\langle 'x \text{ set} \Rightarrow ('x \Rightarrow ('a, 'x)\text{HML\_formula}) \Rightarrow ('a, 'x)\text{HML\_formula} \rangle</math>
  where <math>\langle \text{HML\_weaknor } I \ F = \text{HML\_poss } \tau \ (\text{HML\_conj } I \ (\lambda f. \text{HML\_neg } (F \ f))) \rangle</math>

```

```

definition HML_weaknot ::
  <math>\langle ('a, 'x)\text{HML\_formula} \Rightarrow ('a, 'x)\text{HML\_formula} \rangle</math>
  where <math>\langle \text{HML\_weaknot } \varphi = \text{HML\_weaknor } \{\text{undefined}\} \ (\lambda i. \varphi) \rangle</math>

```

```

inductive_set HML_weak_formulas :: <math>\langle ('a, 'x)\text{HML\_formula} \text{ set} \rangle</math> where
  Base: <math>\langle \text{HML\_true} \in \text{HML\_weak\_formulas} \rangle</math> |
  Obs: <math>\langle \varphi \in \text{HML\_weak\_formulas} \Longrightarrow (\langle \tau \rangle \langle a \rangle \varphi) \in \text{HML\_weak\_formulas} \rangle</math> |
  Conj: <math>\langle (\bigwedge i. i \in I \Longrightarrow F \ i \in \text{HML\_weak\_formulas}) \Longrightarrow \text{HML\_weaknor } I \ F \in \text{HML\_weak\_formulas} \rangle</math>

```

```

lemma weak_backwards_truth:
  assumes
    <math>\langle \varphi \in \text{HML\_weak\_formulas} \rangle</math>
    <math>\langle p \xrightarrow{*} \tau \ p' \rangle</math>
    <math>\langle p' \models \varphi \rangle</math>
  shows
    <math>\langle p \models \varphi \rangle</math>
  <math>\langle \text{proof} \rangle</math>

```

```

lemma tau_a_obs_implies_delay_step:
  assumes <math>\langle p \models \langle \tau \rangle \langle a \rangle \varphi \rangle</math>
  shows <math>\langle \exists p'. p \Rightarrow a \ p' \wedge p' \models \varphi \rangle</math>
  <math>\langle \text{proof} \rangle</math>

```

```

lemma delay_step_implies_tau_a_obs:
  assumes
    <math>\langle p \Rightarrow a \ p' \rangle</math>
    <math>\langle p' \models \varphi \rangle</math>
  shows <math>\langle p \models \langle \tau \rangle \langle a \rangle \varphi \rangle</math>
  <math>\langle \text{proof} \rangle</math>

```

```

end
end

```

## 11 Weak HML and the Contrsimulation Set Game

```

theory Weak_HML_Contrasimulation
  imports
    Contrsim_Set_Game
    HM_Logic_Infinitary
begin

```

## 11.1 Distinguishing Formulas at Winning Attacker Positions

```

locale c_game_with_attacker_strategy =
  c_set_game trans  $\tau$ 
for
  trans :: <'s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool> and
   $\tau$  :: <'a> +
fixes
  strat :: <('s, 'a) c_set_game_node posstrategy> and
  attacker_winning_region :: <('s, 'a) c_set_game_node set> and
  attacker_order
defines
  <attacker_order  $\equiv$  {(g', g). c_set_game_moves g g'  $\wedge$ 
    g  $\in$  attacker_winning_region  $\wedge$  g'  $\in$  attacker_winning_region  $\wedge$ 
    (player1_position g  $\longrightarrow$  g' = strat g)}+>
assumes
  finite_win:
    <wf attacker_order> and
  strat_stays_winning:
    <g  $\in$  attacker_winning_region  $\implies$  player1_position g  $\implies$ 
      strat g  $\in$  attacker_winning_region  $\wedge$  c_set_game_moves g (strat g)> and
  defender_keeps_losing:
    <g  $\in$  attacker_winning_region  $\implies$  c_set_game_defender_node g  $\implies$  c_set_game_moves g g'
       $\implies$  g'  $\in$  attacker_winning_region>
begin

```

This construction of attacker formulas from a game only works if `strat` is a well-founded attacker strategy. (If it's winning and sound, the constructed formula should be distinguishing.)

```

function attack_formula :: <('s, 'a) c_set_game_node  $\Rightarrow$  ('a, 's) HML_formula> where
  <attack_formula (AttackerNode p Q) =
    (if (AttackerNode p Q)  $\in$  attacker_winning_region
      then attack_formula (strat (AttackerNode p Q))
      else HML_true)>
| <attack_formula (DefenderSimNode a p Q) =
  (if (DefenderSimNode a p Q)  $\in$  attacker_winning_region
    then < $\tau$ ><a>(attack_formula (AttackerNode p (dsuccs a Q)))
    else HML_true)>
| <attack_formula (DefenderSwapNode p Q) =
  (if Q = {}  $\vee$  DefenderSwapNode p Q  $\notin$  attacker_winning_region
    then HML_true
    else (HML_weaknor (weak_tau_succs Q)
      ( $\lambda$ q. if q  $\in$  (weak_tau_succs Q)
        then (attack_formula (AttackerNode q {p}))
        else HML_true ))))>
  <proof>

```

```

termination attack_formula
  <proof>

```

```

lemma attacker_defender_switch:
  assumes
    <(AttackerNode p Q)  $\in$  attacker_winning_region>
  shows
    <( $\exists$  a p'. (strat (AttackerNode p Q)) = (DefenderSimNode a p' Q)  $\wedge$  p  $\Rightarrow$  a p'  $\wedge$   $\neg$ tau a)
       $\vee$  ( $\exists$  p'. (strat (AttackerNode p Q)) = (DefenderSwapNode p' Q)  $\wedge$  p  $\mapsto^*$  tau p' )>
  <proof>

```

```

lemma attack_options:

```

```

assumes
  <(AttackerNode p Q) ∈ attacker_winning_region>
shows
  <(∃ a p'. p ⇒ a p' ∧ ¬tau a ∧ strat (AttackerNode p Q) = (DefenderSimNode a p' Q) ∧
    attack_formula (AttackerNode p Q)
    = ⟨τ⟩⟨a⟩(attack_formula (AttackerNode p' (dsuccs a Q))))>
  ∨ (∃ p'. p ⇒* tau p' ∧ strat (AttackerNode p Q) = (DefenderSwapNode p' Q) ∧
    attack_formula (AttackerNode p Q) =
    (HML_weaknor (weak_tau_succs Q) (λq.
      if q ∈ (weak_tau_succs Q)
      then (attack_formula (AttackerNode q {p'}))
      else HML_true )))
  ∨ (Q = {} ∧ attack_formula (AttackerNode p Q) = HML_true)>
<proof>

```

```

lemma distinction_soundness:
  fixes p Q p0 Q0
  defines
    <pQ == AttackerNode p Q>
  defines
    <φ == attack_formula pQ>
  assumes
    <pQ ∈ attacker_winning_region>
  shows
    <p ⊨ φ ∧ (∀ q ∈ Q. ¬ q ⊨ φ)>
<proof>

```

```

lemma distinction_in_language:
  fixes p Q
  defines
    <pQ == AttackerNode p Q>
  defines
    <φ == attack_formula pQ>
  assumes
    <pQ ∈ attacker_winning_region>
  shows
    <φ ∈ HML_weak_formulas>
<proof>

```

end

## 11.2 Attacker Wins on Pairs with Distinguishing Formulas

```

locale c_game_with_attacker_formula =
  c_set_game trans τ
for
  trans :: <'s ⇒ 'a ⇒ 's ⇒ bool> and
  τ :: <'a>
begin

  inductive_set attacker_winning_region :: <('s, 'a) c_set_game_node set> where
    Base: <DefenderSwapNode _ {} ∈ attacker_winning_region> |
    Atk: <(c_set_game_moves (AttackerNode p Q) g' ∧ g' ∈ attacker_winning_region)
      ⇒ (AttackerNode p Q) ∈ attacker_winning_region> |
    Def: <c_set_game_defender_node g ⇒
      (∧ g'. c_set_game_moves g g' ⇒ g' ∈ attacker_winning_region)
      ⇒ g ∈ attacker_winning_region>

```




Figure 2: Example of a  $\tau$ -sink extension with the original transition system in black and the extension in red.

```

lemma attacker_wins_if_defender_set_empty:
  assumes
    <Q = {}>
  shows
    <AttackerNode p Q ∈ attacker_winning_region>
  <proof>

lemma attacker_wr_propagation:
  assumes
    <AttackerNode p' (dsuccs a Q) ∈ attacker_winning_region>
    <p ⇒ a p'>
    <¬tau a>
  shows
    <AttackerNode p Q ∈ attacker_winning_region>
  <proof>

lemma distinction_completeness:
  assumes
    <φ ∈ HML_weak_formulas>
    <distinguishes_from_set φ p Q>
  shows
    <(AttackerNode p Q) ∈ attacker_winning_region>
  <proof>

end
end

```

## 12 Reductions and $\tau$ -sinks

Checking trace inclusion can be reduced to contrasimulation checking, as can weak simulation checking to coupled simulation checking. The trick is to add a  $\tau$ -sink to the transition system, that is, a state that is reachable by  $\tau$ -steps from every other state, and cannot be left. An illustration of such an extension is given in Figure 2. Intuitively, the extension means that the model is allowed to just stop progressing at any point.

We here prove that, on systems with a  $\tau$ -sink, weak similarity equals coupled similarity and weak trace inclusion equals contrasimilarity. We also prove that adding a  $\tau$ -sink to a system does not change weak similarity nor weak trace inclusion relationships within the system. As adding the  $\tau$ -sink only has negligible effect on the system sizes, these facts establish the reducibility relationships.

```

theory Tau_Sinks
imports
  Coupled_Simulation
begin

```

## 12.1 $\tau$ -Sink Properties

```

context lts_tau
begin

```

```

definition tau_sink ::
  <'s  $\Rightarrow$  bool>
where
  <tau_sink p  $\equiv$ 
    ( $\forall a p'. p \mapsto a p' \longrightarrow a = \tau \wedge p = p'$ )  $\wedge$ 
    ( $\forall p0. p0 \mapsto_{\tau} p$ )>

```

The tau sink is a supremum for the weak transition relation.

```

lemma tau_sink_maximal:
  assumes <tau_sink sink>
  shows
    <tau_max sink>
    <(p  $\mapsto^*$  tau sink)>
  <proof>

```

```

lemma sink_has_no_word_transitions:
  assumes
    <tau_sink sink>
    <A  $\neq$  []>
    < $\forall a \in \text{set}(A). a \neq \tau$ >
  shows < $\nexists s'. \text{sink} \Rightarrow_{\$A} s'$ >
  <proof>

```

## 12.2 Contrsimulation Equals Weak Simulation on $\tau$ -Sink Systems

```

lemma sink_coupled_simulates_all_states:
  assumes
    < $\bigwedge p. (p \mapsto^* \text{tau sink})$ >
  shows
    <sink  $\sqsubseteq_{cs}$  p>
  <proof>

```

```

theorem coupled_sim_weak_sim_equiv_on_sink_expansion:
  assumes
    < $\bigwedge p. (p \mapsto^* \text{tau sink})$ >
  shows
    <p  $\sqsubseteq_{ws}$  q  $\longleftrightarrow$  p  $\sqsubseteq_{cs}$  q>
  <proof>

```

## 12.3 Contrsimulation Equals Weak Trace Inclusion on $\tau$ -Sink Systems

```

lemma sink_contrasimulates_all_states:
  fixes A :: "'a list"
  assumes
    <tau_sink sink>
    < $\bigwedge p. (p \mapsto^* \text{tau sink})$ >
  shows

```



```

    <∀ p. sink ⊆c p>
  <proof>

lemma sink_trace_includes_all_states:
  assumes
    <∀ p . (p ⟶* tau sink)>
  shows
    <sink ⊆T p>
  <proof>

lemma trace_incl_with_sink_is_contrasim:
  assumes
    <∀ p . (p ⟶* tau sink)>
    <∀ p . R sink p>
    <trace_inclusion R>
  shows
    <contrasimulation R>
  <proof>

theorem contrasim_trace_incl_equiv_on_sink_expansion_R:
  assumes
    <∀ p . (p ⟶* tau sink)>
    <∀ p . R sink p>
  shows
    <contrasimulation R = trace_inclusion R>
  <proof>

theorem contrasim_trace_incl_equiv_on_sink_expansion:
  assumes
    <∀ p . (p ⟶* tau sink)>
  shows
    <p ⊆T q ⟷ p ⊆c q>
  <proof>

end



## 12.4 Weak Simulation Invariant under $\tau$ -Sink Extension



lemma simulation_tau_sink_1:
  fixes
    step sink R  $\tau$ 
  defines
    <step2 ≡ λ p1 a p2 . (p1 ≠ sink ∧ a =  $\tau$  ∧ p2 = sink) ∨ step p1 a p2>
  assumes
    <∀ a p . ¬ step sink a p>
    <lts_tau.weak_simulation step  $\tau$  R>
  shows
    <lts_tau.weak_simulation step2  $\tau$  (λ p q. p = sink ∨ R p q)>
  <proof>

lemma simulation_tau_sink_2:
  fixes
    step sink R  $\tau$ 
  defines
    <step2 ≡ λ p1 a p2 . (p1 ≠ sink ∧ a =  $\tau$  ∧ p2 = sink) ∨ step p1 a p2>
  assumes
    <∀ a p . ¬ step sink a p ∧ ¬ step p a sink>

```

```

    <lts_tau.weak_simulation step2  $\tau$  ( $\lambda p q. p = \text{sink} \vee R p q$ )>
    < $\wedge p' q' q . (p' = \text{sink} \vee R p' q')$ 
       $\wedge$  lts.steps step2 q (lts_tau.tau  $\tau$ ) q'  $\longrightarrow$  ( $p' = \text{sink} \vee R p' q$ )>
  shows
    <lts_tau.weak_simulation step  $\tau$  ( $\lambda p q. p = \text{sink} \vee R p q$ )>
  <proof>

lemma simulation_sink_invariant:
  fixes
    step sink R  $\tau$ 
  defines
    <step2  $\equiv \lambda p1 a p2 . (p1 \neq \text{sink} \wedge a = \tau \wedge p2 = \text{sink}) \vee \text{step } p1 a p2$ >
  assumes
    < $\wedge a p . \neg \text{step sink a p} \wedge \neg \text{step p a sink}$ >
  shows <lts_tau.weakly_simulated_by step2  $\tau$  p q = lts_tau.weakly_simulated_by step  $\tau$  p q>
  <proof>

```

## 12.5 Trace Inclusion Invariant under $\tau$ -Sink Extension

```

lemma trace_inclusion_sink_invariant:
  fixes
    step sink R  $\tau$ 
  defines
    <step2  $\equiv \lambda p1 a p2 . (p1 \neq \text{sink} \wedge a = \tau \wedge p2 = \text{sink}) \vee \text{step } p1 a p2$ >
  assumes
    < $\wedge a p . \neg \text{step sink a p} \wedge \neg \text{step p a sink}$ >
  shows
    <lts_tau.weakly_trace_included_by step2  $\tau$  p q
      = lts_tau.weakly_trace_included_by step  $\tau$  p q>
  <proof>

end

```

## References

- [1] Bisping, B.: Computing Coupled Similarity. Master’s thesis, Technische Universität Berlin (2018), [https://coupledsim.bbisping.de/bisping\\_computingCoupledSimilarity\\_thesis.pdf](https://coupledsim.bbisping.de/bisping_computingCoupledSimilarity_thesis.pdf)
- [2] Bisping, B., Jansen, D.N.: Linear-time–branching-time spectroscopy accounting for silent steps (2023). <https://doi.org/s10.48550/arXiv.2305.17671>
- [3] Bisping, B., Montanari, L.: A game characterization for contrasimilarity. In: Dardha, O., Castiglioni, V. (eds.) Proceedings Combined 28th International Workshop on Expressiveness in Concurrency and 18th Workshop on Structural Operational Semantics, Electronic Proceedings in Theoretical Computer Science, vol. 339, pp. 27–42. Open Publishing Association (2021). <https://doi.org/10.4204/EPTCS.339.5>
- [4] Bisping, B., Nestmann, U.: Computing coupled similarity. In: Proceedings of TACAS. pp. 244–261. LNCS, Springer (2019). [https://doi.org/10.1007/978-3-030-17462-0\\_14](https://doi.org/10.1007/978-3-030-17462-0_14)
- [5] Bisping, B., Nestmann, U., Peters, K.: Coupled similarity: the first 32 years. Acta Informatica **57**(3–5), 439–463 (2020). <https://doi.org/10.1007/s00236-019-00356-4>
- [6] Fournet, C., Gonthier, G.: A hierarchy of equivalences for asynchronous calculi. The Journal of Logic and Algebraic Programming **63**(1), 131–173 (2005)

- [7] van Glabbeek, R.: A branching time model of CSP. In: Gibson-Robinson, T., Hopcroft, P., Lazić, R. (eds.) *Concurrency, Security, and Puzzles: Essays Dedicated to Andrew William Roscoe on the Occasion of His 60th Birthday*, pp. 272–293. Springer International Publishing, Cham (2017). [https://doi.org/10.1007/978-3-319-51046-0\\_14](https://doi.org/10.1007/978-3-319-51046-0_14)
- [8] van Glabbeek, R.J.: The linear time–branching time spectrum II. In: *International Conference on Concurrency Theory*. pp. 66–81. Springer (1993)
- [9] Parrow, J., Sjödin, P.: Multiway synchronization verified with coupled simulation. In: Cleaveland, W. (ed.) *CONCUR '92: Third International Conference on Concurrency Theory* Stony Brook, NY, USA, August 24–27, 1992 Proceedings. pp. 518–533. Springer Berlin Heidelberg (1992). <https://doi.org/10.1007/BFb0084813>
- [10] Parrow, J., Sjödin, P.: The complete axiomatization of Cs-congruence. In: Enjalbert, P., Mayr, E.W., Wagner, K.W. (eds.) *STACS 94: 11th Annual Symposium on Theoretical Aspects of Computer Science* Caen, France, February 24–26, 1994 Proceedings. pp. 555–568. Springer Berlin Heidelberg (1994). [https://doi.org/10.1007/3-540-57785-8\\_171](https://doi.org/10.1007/3-540-57785-8_171)
- [11] Peters, K., van Glabbeek, R.J.: Analysing and comparing encodability criteria. In: *Proceedings of the Combined 22th International Workshop on Expressiveness in Concurrency and 12th Workshop on Structural Operational Semantics, and 12th Workshop on Structural Operational Semantics, EXPRESS/SOS*. pp. 46–60 (2015). <https://doi.org/10.4204/EPTCS.190.4>
- [12] Peters, K., van Glabbeek, R.J.: Analysing and comparing encodability criteria for process calculi. *Archive of Formal Proofs* (Aug 2015), [http://isa-afp.org/entries/Encodability\\_Process\\_Calculi.html](http://isa-afp.org/entries/Encodability_Process_Calculi.html), Formal proof development
- [13] Pohlmann, M.: Reducing Reactive to Strong Bisimilarity. Bachelor’s thesis, Technische Universität Berlin (2021), <https://maxpohlmann.github.io/Reducing-Reactive-to-Strong-Bisimilarity/>
- [14] Sangiorgi, D.: *Introduction to Bisimulation and Coinduction*. Cambridge University Press, New York, NY, USA (2012)